
Provably Correct Key-Value Maps in Agda
Leveraging Membership to Reach Correctness
Carl Bergman
Johan Henrik Ek
Department of Computer Science and Engineering
Chalmers University of Technology and University of Gothenburg

Abstract
This thesis describes the interface and implementation of a provably correct finite
map in the Agda programming language. Finite maps are fundamental data struc-
tures in computer science, and existing finite maps in Agda can be improved in
terms of correctness and some implementation details. We present an extension of
‘A Theory of Finite Maps’ (Collins & Syme, 1995) and an Agda formalisation of that
extended theory. The formalisation is designed to be structure agnostic, allowing
for many underlying implementations, and its flexibility is demonstrated through a
proof-of-concept simply typed lambda calculus interpreter.

The core of the formalisation is implemented and proven for AVL trees, extending the
Agda standard library with additional proofs and operations. Among these are laws
defining the behaviour of delete and unionWith as well as an alternative unionWith
implementation. The implementation is also differentiated from the standard library
by using erasure to remove, among other things, ordering proofs from the tree data
type and operations. The performance of our implementation is compared to the
standard library and the results show no meaningful performance gains for our
implementation.

Finally, we discuss the inclusion of correct-by-construction membership properties
in the tree structure itself. Such properties have not been introduced in our imple-
mentation. This is due to the constant need to update each proof on every operation
on the collection, since any operation that modifies the structure of the tree may
cause a proof to become irrelevant for the new map.

Keywords: Agda, AVL Trees, Collections, Finite maps

1
Introduction

Collections are data structures that associate keys with values. Common examples
are hash maps or search trees, for example, red-black trees or AVL trees (Adelson-
Velskii & Landis, 1962). Such collections are useful tools for programming and
problem-solving and are often provided directly in a programming language itself or
through libraries.

Agda is a dependently typed functional programming language that is often used
as a proof assistant. The ability to act as a proof assistant makes Agda a viable
option for the formal verification of computer programs. For example, a compiler
or interpreter can be written in such a way that it will always produce code that is
correct according to some language’s specifications. The typing contexts and envir-
onments in compilers and interpreters are commonly implemented using collections
since they provide efficient lookup and insert operations.

Currently, there does not exist any efficient collection for Agda that is proven correct
for all operations. Thus, this thesis aims to design and implement a provably correct
collection for Agda as well as to explore the possibilities of introducing additional
correct-by-construction aspects into the data structure. Furthermore, we will use
proof erasure and evaluate the impact that it has on performance and memory usage.

1.1 Problem and Goal
We wish to explore and implement a provably correct collection in the Agda program-
ming language and proof assistant. To allow for many different implementations,
an abstraction will be defined. The abstraction shall be defined in terms of a set of
operations together with laws for each operation that assert correct behaviour. Our
implementation is guided by the following three (3) questions:

1. Which commonly used functions are lacking proofs and is it possible to add
and prove these?

2. How much of an underlying map implementation can be abstracted while
retaining the usability and flexibility of the map?

3. Can the implementation be written such that parts of the code used only for
verification are erased during compilation?

1



1. Introduction

An initial goal of this thesis was to explore whether correct-by-construction key-
value membership properties could be integrated into the data structure itself. We
found that it was hard to justify the added complexity of development and that there
was no great way of doing it, which is why we removed that goal from the research
questions above. Our thoughts and experiments regarding correct-by-construction
key-value membership can be found in Chapter 6, Section 6.1.

1.2 Limitations
The implementation of this thesis is based on AVL trees, following previous work
in the Agda standard library (The Agda Community, 2023). Only AVL trees have
been proven correct for this thesis, so there might be other data structures that are
more suited for the task at hand. An alternative that is explored to some degree in
Chapter 6 is weight-balanced trees, but none of their properties have been proven.

The thesis focuses on a few operations, namely insertion, lookup, deletion, and
union, along with membership data types. This limitation of scope is useful as
it allows for a focused review of existing and new laws. Certain aspects of the
implementation have also been restricted to allow for ease of proving. An example
of this is StrictTotalOrder. The StrictTotalOrder from the standard library allows any
instance of equality to be defined. Our implementation uses the built-in equality
relation, leading to less generality and potential issues in some data types.

1.3 Ethical and Ecological Implications
Faulty software can have dire consequences, as in Ben-Ari (2001) and Wong et al.
(2009). Agda is an open source language that strives for complete correctness as it
is used in formal verification and mathematical reasoning. Maintaining correctness
reduces the possibility of errors caused by the language itself, but a poorly written
program might still produce incorrect or inconsistent results. By providing a com-
monly used provably correct data structure, we assist Agda developers in writing
software with fewer bugs and errors. Additionally, Agda’s open-source nature al-
lows these improvements in correctness to be distributed to a large group of users,
resulting in a greater impact on the programming community as a whole.

It is difficult to find any possible ecological implications for our research as the
connection between the correctness of software and its ecological impact is a tenuous
one at best.

1.4 Code Availability
The complete source code and programs used for the benchmarking are publicly avail-
able in the following GitHub repository: https://github.com/MiztaOak/Provably_
Correct_Finite_Maps.

1. Introduction

1.5 Related Work
Collections are common data structures, and therefore there exist many different
implementations. In this section, we will delve into existing collections and literature
related to this work.

1.5.1 Collections
Conceptually, a collection can be defined as a “collection of things”, meaning some-
thing that keeps data. Collections do not require the data to be associative, so a
collection can be as a set, a list, or anything else that can store a variable amount of
data. However, it is considerably common to store associative data in collections. A
focus of this thesis is the membership aspect, thus collections that store associative
data are of primary focus, but the theory and abstraction can still be used and
adapted for non-associative collection types.

One type of collection that keeps associative data is finite maps. The data stored in
finite maps are usually referred to as “keys” that are associated with “values”. One
theory of finite maps is presented by Collins and Syme (1995). In this theory a finite
map consists of a set of functions together with laws that define the behaviour of
the functions, and thereby the map.

The available literature on the theory of finite maps or collections appears to be
rather limited. However, we believe that the theory presented by Collins and Syme
is still relevant based on its similarity to finite maps in existing libraries, e.g., Coq’s
standard library (The Coq Team, 1989) and Agda’s standard library (The Agda
Community, 2023). For that reason, it is used as the basis for the theory presented
in this thesis.

1.5.2 Representations of Collections
Collections have many applications in computer science, for example, graphs, set
representations, databases, caching, and many more. We place special importance
on the usage of collections, specifically finite maps, for environment and context
representations of programming languages. To our knowledge, the usage of data
structures to maintain environments was first shown by McCarthy for usage in the
LISP programming system (McCarthy, 1960).

The implementation described in McCarthy’s classic paper uses a linked list to model
the collection. Modern collections frequently use other data structures internally,
most commonly tree structures. This is in part due to the increased efficiency of the
operations, but also because in naïve implementations two lists might not be logically
equal despite behaving in the same way when treating them as finite maps (Collins
& Syme, 1995).

It is also possible to implement a collection using an array structure. To assert
that each key is unique in an array implementation, an excellent hash function is
required, unless a very large array is used. The first of which might be difficult to

3



1. Introduction

implement or expensive to execute, and the second requires a lot of memory. Tree
structures inherently ensure that each key is unique and also minimise memory usage
by allocating space only for the keys and values that are actually utilised.

Listing 1.1: An elementary function-based collection in Agda.

module FunMap (K : Set) (V : Set) (compare : K → K → Dec K) where

Map : Set
Map = K → Maybe V

insert : K → V → Map → Map
insert k v m c with compare k c
... | yes refl = just v
... | no _ = nothing

lookup : K → Map → Maybe V
lookup k m = m k

delete : K → Map → Map
delete k m c with compare k c
... | yes refl = nothing
... | no _ = m c

Arguably, the most intuitive and elegant definition of key-value maps is a function-
based approach, like Collins and Syme (1995), or the example shown in Listing 1.1.
However, in practice, this can be less efficient than a tree-based approach, as com-
posed functions essentially construct a linked list, causing common operations to
have linear complexity compared to logarithmic complexity in trees.

There are many different tree structures for representing collections. While it is
possible to use simple search trees it is more appropriate to use balanced search
trees as the worst-case time complexity of most operations is improved. The Agda
standard library uses AVL trees (Adelson-Velskii & Landis, 1962) to represent a
finite map.

AVL trees are balanced by the height of the children. Each child may not differ
by more than one as per the AVL invariant. Such a balancing causes the trees to
achieve logarithmic complexity in terms of size for the essential operations, insert,
lookup, and delete (Adelson-Velskii & Landis, 1962; Knuth, 1998).

Adams describes an alternate approach with weight-balanced trees (Adams, 1993). A
weight-balanced tree is balanced by the size of the children and some constant factor
or ratio to create a balance. According to Adams, the benefit of balancing by size
compared to height is that the size can be utilised to create “smarter” constructors.
A simple, or naïve, constructor constructs a tree by concatenating two subtrees 𝑡1
and 𝑡2 at a root node and assigning the size of the new tree to be 1 + size(𝑡1) +

4



1. Introduction

size(𝑡2). A “smart” constructor would check and confirm that the balancing of the
new tree is correct, and if not, rebalance it. Smart constructors greatly simplify his
algorithms, as all rotations are performed in one place. Adams smart constructors
can be likened to the join function of Blelloch et al. (2016), despite him providing
join with a different name in concat3. This is because Blelloch et al. show that join
can be used for the same purposes with helping functions splitAt and join2.

Different function implementations can vary in efficiency; an example of this is the
union function in the Agda standard library. The existing implementation uses a
fold-based approach, meaning that it inserts every element of one map into another.
The complexity of such an algorithm is defined by the number of elements in the
two trees. Inserting an element into a tree of size 𝑚 is 𝑂(log𝑚) (Knuth, 1998) and
performing this action 𝑛 times, as done for a tree of size 𝑛, results in a complexity of
𝑂(𝑛 log𝑚). More efficient approaches are, for example, the hedge union algorithm
from Adams (1993). Hedge union works by splitting one of the trees at the root
node of the other and then joining them together. Blelloch et al. (2016) show that
this algorithm has a time complexity of 𝑂(𝑚 log( 𝑛

𝑚 + 1)) for two trees with heights
𝑚 and 𝑛 such that 𝑚 ≤ 𝑛. In addition to an overall improved time complexity,
the run time is generally more consistent given different input trees. This is easily
recognisable by analysing the two complexities or the functions themselves, but is
also shown by us in Chapter 5 as a part of our performance evaluation. The union
implementation of this thesis is based on the formalisation of Adams’ hedge union
algorithm by Blelloch et al. (2016).

1.5.3 Verified Collections
Melkonian (2024) has written an Agda library that contains an abstraction for fi-
nite maps. The finite map abstraction defines a set of functions and laws similar to
Collins and Syme but is extended with several functions and definitions. In addition
to an abstraction, there is an example of a function-based finite map in the library.
The map definition in this library has been a valuable resource and source of inspir-
ation for this project. The AVL based finite map of the standard library was used
as a basis for our work as it already exists in the target language and is integrated
in the Agda ecosystem. Therefore, we hope that some of the work in this thesis can
be used by the extended Agda community in the future.

Libraries of languages similar to Agda, e.g., Coq (The Coq Team, 1989) or Isa-
belle (Nipkow et al., 2002) also have finite maps with different implementations.
The AVL map in the Coq standard library is similar to the work by Adams (1993)
in that most of the balancing is taken care of by a function similar to a smart
constructor and by the join function. The approach of Isabelle is very similar to
the Agda standard library’s AVL trees, both of which are correct by construction
directly in terms of the balancing instead of the heights. This means a data type
that represents the balance factor, i.e., either both children are of the same height
or they differ by one in some direction.

Neither Coq nor Isabelle is correct-by-construction in terms of ordering. McBride
(2014) shows in his paper ‘How to keep your neighbours in order’ an elegant technique

5



1. Introduction

to define ordering correctness in the tree itself, this technique is described in detail
in Section 2.3 as it is used in our implementation. Ordering correctness à la McBride
is also used in the AVL tree of the Agda standard library. Coq, like Agda, is based
on the calculus of constructions (Coquand & Huet, 1988) whereas Isabelle is a more
generic proof assistant. The proofs observed in Isabelle are in higher-order logic.
While Coq and Isabelle contains proven map implementations, we are not able to
use their proofs other than for brief analysis and inspiration. The reason for this is
that Isabelle and Coq tend to utilise automation in proving, whereas Agda focuses
on a more direct approach where the developer manually writes the proofs. In short,
the more automated proving process based on different proving tactics of Coq and
Isabelle makes it hard to use their results in our proofs.

6



2
Theory

To realise the goal of defining and exploring a provably correct finite map, we must
first learn more about integral concepts relating to it. This chapter introduces
the most important concepts at a high level. These being the Agda programming
language, correct-by-construction techniques, and finite maps.

2.1 Agda
Agda (Agda Developers, 2024) is an actively developed functional programming
language. Notable features of Agda are its dependent type system and termination
checking. To run Agda code it is necessary to first translate it into another language,
e.g., Haskell by using the MAlonzo compiler (Benke, 2007), and then compile the
resulting source code. Dependent typing allows for types to depend on values, and
using this in clever ways allows the developer to express invariants directly in the
data structure. Such a type could, for example, be the type of all integers strictly
larger than some specified integer or a list that is sorted by definition. The syntax
of Agda is similar to that of Haskell, but it has some differences that are valuable
to know.

• Agda identifiers can contain operators and arbitrary unicode. This allows for
usage of mathematical notation, e.g., ≡ for equivalence or ∈ for membership.

• All functions are required to be total. An example of such a function is the
commonly used head function that returns the first element in a list. In Haskell,
one can simply pattern match on the head of the list and return it; but if the
list is empty, this strategy does not execute successfully. In Agda the head
function must either result in a Maybe answer, giving nothing if the value does
not exist, or it can use the dependent type system to express in the type
declaration that the list cannot be empty.

• When developing Agda programs it is common to only utilise the type- and
termination checkers, meaning that code is not necessarily executed at any
point of development. If the developer wants to type check their program but
has yet to finish a long proof, it is possible to use a so called “hole” in place
of the remaining proof. Holes are denoted with ? or {!!} . Anything in a hole,
e.g., {!m!} , is not type checked, but it is possible to use Agda to look at m’s
type and the surrounding context.

7



2. Theory

• It is possible to declare an argument as implicit if the type checker can figure
it out itself. This is done by declaring a type or variable in single braces. For
example:

n≡n⟹1+n≡1+n : {n : ℕ} → n ≡ n → suc n ≡ suc n
n≡n⟹1+n≡1+n refl = refl

In this case, it is enough to provide a proof of equality since n can be derived
from that proof.

• Instance arguments can be declared using double braces, e.g.,
{{x : Type }}, where Type is some record, data type, or variable. Instance ar-
guments can be thought of as instances of type-class declarations. An example
from Agda’s documentation:

_==_ : {A : Set} → {{eqA : Eq A}} → A → A → Bool

In the example Eq A is some appropriate type that allows for elements of A
to be compared.

• A record is a type that is used to group values together. A field in a record
declaration contains type declarations associated with the record. Within a
record one can declare a constructor for the creation of record values. In a
record value a field will instead be associated with a value of the declared
type. For example, the dependent pair can be defined using a record:

record Σ (S : Set) (T : S → Set) : Set where
constructor _,_
field fst : S

snd : T fst

The Σ type will be used throughout the code and represents the type where the
type of the second element depends on the first. This type is most commonly
seen in our implementation using the _×_ function:

_×_ : (S T : Set) → Set
S × T = Σ S 𝜆 _ → T
infixr 5 _×_

• The --erasure option will be used in later chapters. With this option enabled,
one can specify arguments, types, or functions with @0 or @erased causing
them to be erased, or ignored, during compilation so that they are absent at
run-time. This can, for example, remove an expensive proof from an otherwise
efficient operation.

• Agda allows for local modules, meaning that module can be declared almost at
any point in a file. This is commonly used to define module-specific parameters
required by the data types, functions, and records in the module. An example
of this can be seen in Section 2.3 later in this chapter. There, the BST type
exists within the BSTM module which is parameterised by the relation R.

One of the key features of Agda is its ability to act as a proof assistant. This works
because of the Curry-Howard correspondence (Howard et al., 1980), which expresses
an equivalence between purely functional computer programs and mathematical
proofs. In particular, a type corresponds to a proposition and a term to a proof. In
the context of a language like Agda, this would mean that the type of a function
represents a proposition, and its implementation the proof. For example, given a
data type representing equivalence, we can show that 1 + 1 ≡ 2:

data ℕ : Set where
zero : ℕ
suc : ℕ → ℕ

_+_ : ℕ → ℕ → ℕ
a + zero = a
a + (suc b) = suc (a + b)

data _≡_ {A : Set} (n : A) : A → Set where
refl : n ≡ n

1+1≡2 : (suc zero) + (suc zero) ≡ suc (suc zero)
1+1≡2 = refl

Further, in the proof that 1 + 1 ≡ 2, Agda immediately understands that the type
and proposition, (suc zero) + (suc zero) ≡ suc (suc zero) is equivalent, as it uses
the definition of the _+_ function first to simplify the expression. This allows us
to immediately use the constructor of _≡_, refl, as the implementation and proof.
More involved proofs may require some assistance from the developer, for example:

_⊔_ : ℕ → ℕ → ℕ
zero ⊔ n = n
suc a ⊔ zero = suc a
suc a ⊔ suc b = suc (a ⊔ b)

a⊔a≡a : (a : ℕ) → a ⊔ a ≡ a
a⊔a≡a zero = refl
a⊔a≡a (suc a) rewrite a⊔a≡a a = refl

The _⊔_ operation expresses the maximum of two numbers, and the proof shows
that the maximum of a number compared with itself is, in fact, the same number.
The issue is that when normalising, the final line in _⊔_, suc (a ⊔ a), cannot simplify
to suc a. The rewrite call helps Agda by showing that a ⊔ a ≡ a, which in turn
causes the proof to look for suc a ≡ suc a instead of suc (a ⊔ a).

2.2 A Theory of Finite Maps
As previously mentioned collections, also known as finite maps, associative lists,
or key-value maps, are well-known and often used data structures within software

9

Table 2.1: Functions from Collins and Syme (1995).

Empty The finite map with no elements in its domain.
Update 𝑓 (𝑥, 𝑦) Map 𝑥 to 𝑦 in the finite map 𝑓 . There is no restriction on

whether 𝑥 exists in 𝑓 , if 𝑥 ∈ 𝑓 then the mapping will be
updated so that 𝑥 maps to 𝑦 in 𝑓 .

Apply 𝑓 𝑥 If 𝑥 ∈ 𝑓 then Apply 𝑓 𝑥 returns some value 𝑣 such that 𝑥
maps to 𝑣 in 𝑓 .

Domain 𝑓 𝑥 Return a boolean value of True if 𝑥 ∈ 𝑓 , if not, return False.

development. Arguably one of the more important use cases for collections is the
modeling of typing contexts and environments in programming languages. In this
section, we wish to dive deeper into the theory of such data structures.

A finite map may have different purposes, but essential to most, if not for all applic-
ations, is the ability to associate keys with values. A theory of finite maps contains
two parts: the functions that act upon a finite map and the axioms that define their
behaviour. One such theory is described by Collins and Syme (1995). The functions
described in their theory are available in Table 2.1.

For anyone familiar with computer science or programming, these functions are
familiar, albeit some with different names. Update would be referred to as insert,
Apply as lookup, and Domain as member. Collins and Syme explain that this simple
set of functions is enough to be used to prove any property of finite maps. The set
of functions is not provided on its own; each function must have a corresponding set
of axioms to assert that the behaviour is correct. An axiom, in essence, is simply
a proof of a certain behaviour of some function, but they are referred to as axioms
as Collins and Syme reasons that they must all exist together, and if any axiom is
removed the remaining set can no longer define a complete theory.

⊢ ∀𝑓 𝑎 𝑏. Apply (Update 𝑓 (𝑎, 𝑏)) 𝑎 = 𝑏 (2.1)
⊢ ∀𝑥 𝑎. (𝑎 ≠ 𝑐) → (2.2)

∀𝑓 𝑏 𝑑. (Update (Update 𝑓 (𝑎, 𝑏)))(𝑐, 𝑑) = Update (Update 𝑓 (𝑐, 𝑑))(𝑎, 𝑏)
⊢ ∀𝑓 𝑎 𝑏 𝑐. Update(Update 𝑓(𝑎, 𝑏))(𝑎, 𝑐) = Update 𝑓 (𝑎, 𝑐) (2.3)
⊢ ∀𝑎.¬(Domain Empty 𝑎) (2.4)
⊢ ∀𝑓 𝑎 𝑏 𝑥. Domain (Update 𝑓 (𝑎, 𝑏))𝑥 = (𝑥 = 𝑎) ∨ Domain 𝑓 𝑥 (2.5)
⊢ ∀𝑃 . (2.6)

𝑃 Empty ∧
(∀𝑓. 𝑃 𝑓) → (∀𝑥.¬(Domain 𝑓 𝑥) → ∀𝑦. 𝑃 (Update 𝑓 (𝑥, 𝑦)))
→ ∀𝑓. 𝑃 𝑓

Figure 2.1: The axioms of Collins and Syme (1995).

The first six axioms shown in figure 2.1 express the intended behaviour of the func-

tions, with axiom 2.6 being an inductive principle. Specifically, it is a strong induc-
tion principle.

Collins and Syme (1995) have proven all of their axioms in the HOL proof assistant,
thereby showing that their theory is correct. They also show that the theory is
consistent, as they are able to derive it using a function-based model. Finally, they
show that the theory is complete, by utilising the same function-based model and
showing that a bijection exists between the map described by the theory and the
model.

2.3 Correct-by-Construction
Correct-by-construction is a technique used to assert correctness within the data
structures themselves. This technique is especially useful for dependently typed
languages, as the types can depend on instances of types. This allows correct-by-
construction data structures to, at times, immediately be used to prove the correct-
ness of operations, making it a matter of type-checking (Pierce, 2002).

The paper ‘How to keep your neighbours in order’ (McBride, 2014) describes a
general technique to assert the correctness of ordering in data structures where an
ordering invariant is necessary. The technique describes how to leverage Agda to
automatically infer proofs of correctness while also hiding these proofs themselves,
allowing for readable code and ease of use for the developer. McBride describes two
different instances of the technique: for a universe of the PolyP language extension
by Jansson and Jeuring (1997) and binary search trees.

Appropriately, we can show a type of trees from the paper that is ordered by con-
struction as an example. The following code originates from McBride’s paper, is
then further modified by our supervisor Andreas Abel, and finally modified again
and presented here by us:

Rel : Set → Set1
Rel A = A × A → Set

data Order (A : Set) : Set where
top : Order A
# : A → Order A
bot : Order A

ext : ∀ {A} → Rel A → Rel (Order A)
ext R (_ , top) = ⊤
ext R (# x , # y) = R (x , y)
ext R (bot , _) = ⊤
ext R _ = ⊥

module BSTM {A : Set} (R : Rel A) where

data BST (l u : Order A) : Set where

11

leaf : {{l<u : ext R (l , u)}} → BST l u
node : (k : A) → BST l (# k) → BST (# k) u → BST l u

Along with the tree definition, BST, there are some auxiliary definitions, the Order
data type and the ext function. The ext function expresses the ordering of the Order
data type by utilising the relation on A. The constructors top and bot are shown to
represent the top and bottom of an ordering. The data type itself is parameterised
using the ordering, l and u in the BST type. The crucial part is in the left and right
tree in the node constructor of BST, where we can see the # constructor of Order in
action. The proof information of the ordering is stored in the leaves.

So how does this express an ordering? The simple and elegant answer is that an
initial node is constructed such that both leaves contain evidence that the element
is larger than bot and smaller than top:

1

bot ≤ (#1) (#1) ≤ top

Insertion of a new element into this tree, e.g., 2, results in the following tree:

1

bot ≤ (#1) 2

(#1) ≤ (#2) (#2) ≤ top

See where this is going? Inserting one more element, 0, will make us certain:

1

0

bot ≤ (#0) (#0) ≤ (#1)

2

(#1) ≤ (#2) (#2) ≤ top

The ordering proofs stored in the leaves express exactly the ordering of the tree!
The technique described is used for McBride’s BST, the AVL tree of the standard
library, and our AVL-tree implementation.

3
An Extension of A Theory of

Finite Maps

The theory presented by Collins and Syme (1995) provides a solid foundation for
any implementation of collections. Collins and Syme uses the word “axiom” when
defining the behaviour of finite map operations, we will instead use the term “law”.
The reason for this is that the laws defined by us are not created just for specification
of a theory but also with Agda development in mind. Additional functions and data
types require that new laws be added and existing laws be modified. In this chapter,
we present and reason about the additional functions and laws presented in our
theory. In essence, our theory is grounded in Collins and Syme’s theory and is
extended by two functions and two data types along with their laws.

The functions were gathered by an analysis of existing libraries as well as personal
experience in programming with collections. Added functions are insertWith, de-
lete, unionWith, and a lookup function related to membership types. A reason for
deciding on the unionWith function is the possibility to reuse some of the helper
functions; this is covered in depth in Chapter 6.

Instead of providing the Domain function, we decided to use membership types in-
stead. There are two data types defined, _ ∈ _ and _ ↦ _ ∈ _. The two types are
provided to express different, but similar, properties. These data types are expressed
by following two types:

_ ∈ _ ∶ 𝐾𝑒𝑦 → 𝑀𝑎𝑝 → 𝑆𝑒𝑡
_ ↦ _ ∈ _ ∶ 𝐾𝑒𝑦 → 𝑉 𝑎𝑙𝑢𝑒 → 𝑀𝑎𝑝 → 𝑆𝑒𝑡

The type 𝑘 ∈ 𝑚 expresses that a key 𝑘 exists in a collection 𝑚, and 𝑘 ↦ 𝑣 ∈ 𝑚
represents a more exact relationship where 𝑘 is associated with 𝑣 in the collection 𝑚.
There exists an equivalence between 𝑘 ↦ 𝑣 ∈ 𝑚 and lookup 𝑘 𝑚 ≡ 𝑣, an interested
reader may look at the laws in Figure 3.1 where it is shown.

The decision of using data types is to facilitate the verification of membership prop-
erties, because the data type expresses an absolute relationship. Compare this to
Domain and Apply where one would still need to prove that Domain 𝑘 𝑚 ≡ True or
Apply 𝑘 𝑚 ≡ 𝑣 for the key 𝑘, collection 𝑚, and value 𝑣. In such cases, we believe that
using the data types that express membership is easier to reason with compared to
first showing that Domain 𝑘 𝑚 ≡ True or Apply 𝑘 𝑚 ≡ 𝑣.

3. An Extension of A Theory of Finite Maps

⊢ ∀𝑘 𝑣 → ¬(𝑘 ↦ 𝑣 ∈ ∅) (3.1)
⊢ ∀𝑘 → 𝑘 ∉ ∅ (3.2)
⊢ ∀𝑘 → lookup ∅ 𝑘 ≡ nothing (3.3)
⊢ ∀𝑚 𝑘 {𝑣} → lookup 𝑚 𝑘 ≡ just 𝑣 → 𝑘 ↦ 𝑣 ∈ 𝑚 (3.4)
⊢ ∀𝑚 𝑘 𝑣 → 𝑘 ↦ 𝑣 ∈ 𝑚 → lookup 𝑚 𝑘 ≡ just 𝑣 (3.5)
⊢ 𝑚 𝑘 → 𝑘 ∉ 𝑚 → lookup 𝑚 𝑘 ≡ nothing (3.6)
⊢ ∀𝑚 𝑘 → lookup 𝑚 𝑘 ≡ nothing → 𝑘 ∉ 𝑚 (3.7)
⊢ ∀𝑚 𝑘 𝑣 → 𝑘 ↦ 𝑣 ∈ 𝑚 → 𝑘 ∈ 𝑚 (3.8)
⊢ ∀𝑚 𝑘 → 𝑘 ∈ 𝑚 → (∃𝑣 → 𝑘 ↦ 𝑣 ∈ 𝑚) (3.9)

Figure 3.1: Laws defining the behaviour of lookup, _ ↦ _ ∈ _, _ ∈ _, as well as
the relation between them.

⊢ ∀𝑘 𝑚 𝑓 → lookup 𝑘 (insertWith 𝑘 𝑓 𝑚) ≡ 𝑓 (lookup 𝑚 𝑘) (3.10)
⊢ ∀𝑘 𝑘′ 𝑓 𝑓 ′ 𝑚 → 𝑘 ≢ 𝑘 (3.11)

→ insertWith 𝑘 𝑓 (insertWith 𝑘′ 𝑓 ′ 𝑚)
≐ insertWith 𝑘′ 𝑓 ′ (insertWith 𝑘 𝑓 𝑚)

⊢ ∀𝑚 𝑘 𝑥 𝑣 𝑓 → 𝑥 ↦ 𝑣 ∈ (insertWith 𝑘 𝑓 𝑚) → (𝑥 ≡ 𝑘) ⊍ 𝑥 ↦ 𝑣 ∈ 𝑚
(3.12)

⊢ ∀𝑘 𝑣 𝑚 → 𝑘 ↦ 𝑣 ∈ (insert 𝑘 𝑣 𝑚) (3.13)
⊢ ∀𝑘 𝑘′ 𝑣 𝑣′ 𝑚 → 𝑘 ↦ 𝑣 ∈ 𝑚 → 𝑘 ≢ 𝑘′ (3.14)

→ 𝑘 ↦ 𝑣 ∈ (insert 𝑘′ 𝑣′ 𝑚)
⊢ ∀𝑘 𝑣 𝑣′ 𝑚 → 𝑘 ↦ 𝑣 ∈ (insert 𝑘 𝑣′ 𝑚) → 𝑣 ≡ 𝑣′ (3.15)

Figure 3.2: Insert related laws.

⊢ 𝑃 (3.16)
→ 𝑃 ∅ × (∀𝑚 → 𝑃 𝑚 → ∀𝑘 𝑣 → 𝑘 ∉ 𝑚 → 𝑃 (insert 𝑘 𝑣 𝑚))
→ (∀𝑚 → 𝑃𝑚)

⊢ 𝑃 (3.17)
→ 𝑃 ∅ × (∀𝑚 → 𝑃 𝑚 → ∀𝑘 𝑣 → 𝑃 (insert 𝑘 𝑣 𝑚))
→ (∀𝑚 → 𝑃 𝑚)

Figure 3.3: The strong and weak induction principles for finite maps, respectively.

3.1 Modification of Collins and Symes’ Axioms
All functions except for Domain in Collins and Syme (1995) exist in our theory, but,
as mentioned previously, the names differ. The functions used by Collins and Syme
are visible in Table 2.1 in the previous chapter. The names are changed to more
widely recognised names, for example insert instead of Update and lookup instead
of Apply. Additionally, we utilise Agda’s unicode support to express the empty set
as ∅ and the membership data types using appropriate symbols, i.e., ∈ and ↦.

Collins and Symes’ set of laws is extended with seven (7) new laws, visible in Figure
3.4. These laws cover the expected behaviour of lookup, insert, and insertWith
as well as some interactions between insertWith and lookup. The reasoning behind
introducing new laws is partly due to the new data types that express membership
and also because they assist in proof writing and reasoning. The complete set of
laws for the basic set of operations is available in Figures 3.1, 3.2, and 3.3.

⊢ ∀𝑘 → lookup 𝑘 ∅ ≡ nothing
⊢ ∀𝑚 𝑘 𝑣 → lookup 𝑘 𝑚 ≡ 𝑣 → 𝑘 ↦ 𝑣 ∈ 𝑚
⊢ ∀𝑚 𝑘 𝑣 → 𝑘 ↦ 𝑣 ∈ 𝑚 → lookup 𝑘 𝑚 ≡ 𝑣
⊢ ∀𝑘 𝑚 𝑓 → lookup 𝑘 (insertWith 𝑘 𝑓 𝑚) ≡ 𝑓 (lookup 𝑚 𝑘)
⊢ ∀𝑘 𝑣 𝑚 → 𝑘 ↦ 𝑣 ∈ (insert 𝑘 𝑣 𝑚)
⊢ ∀𝑘 𝑣 𝑣′ 𝑚 → 𝑘 ↦ 𝑣 ∈ (insert 𝑘 𝑣′ 𝑚) → 𝑣 ≡ 𝑣′
⊢ ∀𝑘 𝑘′ 𝑣 𝑣′ 𝑚 → 𝑘 ↦ 𝑣 ∈ 𝑚 → 𝑘 ≢ 𝑘′

→ 𝑘 ↦ 𝑣 ∈ (insert 𝑘′ 𝑣′𝑚)

Figure 3.4: Additional insertion and lookup laws introduced by us.

3.2 Equivalence Between Finite Maps
The notion of equality is important for the next section of this chapter. Previously,
we have seen the ≡ notation, but this is not suitable when working with all imple-
mentations of finite maps. The definition of ≡ in Agda defines intensional equality,
meaning that the structures themselves must also be equal. For example, consider
the following two tree representations of a map:

2 ↦ a

1 ↦ b

1 ↦ b

2 ↦ a

As we can see the domain of the two trees are the same and the same elements map
to the same values, but as the structures are not equivalent they are not equal by
≡ in Agda. For every element to exist in both maps and map to the same values

is instead defined to be equality between maps, and it is the definition that we use.
The equality between maps is represented by the symbol ≐ in our theory and is
defined as follows:

𝑚1 ≐ 𝑚2 = 𝑚1 ⊆ 𝑚2 ×𝑚2 ⊆ 𝑚1.
As part of the definition we use ⊆, which is defined for 𝑡1 ⊆ 𝑡2 when all elements in
the domain of 𝑡1 exist in 𝑡2 and are mapped to the same values:

𝑡1 ⊆ 𝑡2 = ∀𝑘 𝑣. 𝑘 ↦ 𝑣 ∈ 𝑡1 → 𝑘 ↦ 𝑣 ∈ 𝑡2.

3.3 Additional Functions and their Laws

⊢ ∀𝑘 𝑚 → 𝑘 ∉ 𝑚 → delete 𝑘 𝑚 ≐ 𝑚 (3.18)
⊢ ∀𝑘 𝑚 → 𝑘 ∉ delete 𝑘 𝑚 (3.19)
⊢ ∀𝑘1 𝑘2 𝑣 𝑚 → 𝑘1 ↦ 𝑣 ∈ 𝑚 → 𝑘1 ≢ 𝑘2 → 𝑘1 ↦ 𝑣 ∈ delete 𝑘2 𝑚 (3.20)
⊢ ∀𝑘1 𝑘2 𝑣 𝑚 → 𝑘1 ↦ 𝑣 ∈ delete 𝑘2 𝑚 → 𝑘1 ↦ 𝑣 ∈ 𝑚 (3.21)
⊢ ∀𝑘 𝑣 𝑚 → 𝑘 ↦ 𝑣 ∉ delete 𝑘 𝑚 (3.22)
⊢ ∀𝑘1 𝑘2 𝑚 → delete 𝑘1 (delete 𝑘2 𝑚) ≐ delete 𝑘2 (delete 𝑘1 𝑚) (3.23)

Figure 3.5: Laws related to delete.

As mentioned in the introduction of this chapter, we extend the theory with two
functions, namely delete and unionWith. The behaviour of these functions are
expressed by the laws in Figures 3.5 and 3.6. In order to make sense of the behaviour
defined by the laws, we may also express it in simpler terms.

delete has the sole purpose of deleting an element, and its associated value, from a
collection. A deleted key in a collection m must not map to any value in m, and, it
must not exist in m. The deletion laws, shown in Figure 3.5, express these properties
as well as commutativity.

The purpose of unionWith is to “merge” two collections using a function to determine
what happens in case of a collision. The behaviour of keys in a merged map is defined
by laws 3.26 and 3.27. Together, they state that every key in the input maps must
exist in the resulting map and that every key in the union must exist in the first
input map, or the second input map, or in both. Laws 3.26 and 3.27 must both
exist in order to assert that no additional keys are added or removed from a union
of two maps.

However, defining a law for the value associated with a key is more difficult. This is
because any union implementation must be biased in some way since the comparison
and conflict resolution function will be performed in some order depending on the
union call. We propose law 3.28 for our implementation, but this law may require
modification for other implementations with different biases. In addition to the laws
expressing key and value behaviour, there is a law of congruence 3.31.

⊢ ∀𝑚 𝑓 → unionWith 𝑓 𝑚 ∅ ≐ 𝑚 (3.24)
⊢ ∀𝑚 𝑓 → unionWith 𝑓 ∅ 𝑚 ≐ 𝑚 (3.25)
⊢ ∀𝑚1 𝑚2 𝑓 𝑘 (3.26)

→ 𝑘 ∈ unionWith 𝑓 𝑚1 𝑚2
→ 𝑘 ∈ 𝑚1 ⊍ 𝑘 ∈ 𝑚2 ⊍ (𝑘 ∈ 𝑚1 × 𝑘 ∈ 𝑚2)

⊢ ∀𝑚1 𝑚2 𝑓 𝑘 (3.27)
→ 𝑘 ∈ 𝑚1 ⊍ 𝑘 ∈ 𝑚2
→ 𝑘 ∈ unionWith 𝑓 𝑚1 𝑚2

⊢ ∀𝑘 𝑣1 𝑣2 𝑚1 𝑚2 𝑓 (3.28)
→ 𝑘 ↦ 𝑣1 ∈ 𝑚1
→ 𝑘 ↦ 𝑣2 ∈ 𝑚2
→ 𝑘 ↦ 𝑓 𝑣1 𝑣2 ∈ unionWith 𝑓 𝑚1 𝑚2

⊢ ∀𝑘 𝑣 𝑚1 𝑚2 𝑓 (3.29)
→ 𝑘 ↦ 𝑣 ∈ 𝑚1
→ 𝑘 ∉ 𝑚2
→ 𝑘 ↦ 𝑣 ∈ unionWith 𝑓 𝑚1 𝑚2

⊢ ∀𝑘 𝑣 𝑚1 𝑚2 𝑓 (3.30)
→ 𝑘 ∉ 𝑚1
→ 𝑘 ↦ 𝑣 ∈ 𝑚2
→ 𝑘 ↦ 𝑣 ∈ unionWith 𝑓 𝑚1 𝑚2

⊢ ∀𝑓 𝑚1 𝑚2 𝑚3 (3.31)
→ 𝑚1 ≐ 𝑚2
→ unionWith 𝑓 𝑚1 𝑚3 ≐ unionWith 𝑓 𝑚2 𝑚3

Figure 3.6: Laws related to unionWith.

4
Implementation

This chapter is dedicated to presenting our finite map implementation in Agda. It
contains three parts; an abstract representation, the underlying implementation,
and finally the implementation of the abstraction.

4.1 Abstraction
The extended theory presented in Chapter 3 is implemented in Agda as the following
three (3) interfaces:

• BasicMap, replicating the extended theory of Collins and Syme described in
Section 3.1.

• DeletableMap, extending a Basicmap by adding delete and the laws in Fig-
ure 3.5.

• MergableMap, which extends a Basicmap by adding union and the correspond-
ing laws in Figure 3.6.

The splitting of the theory into three separate interfaces allows map implementations
without a union or deletion function to reason about the correctness of insertion and
lookup in line with Collins and Syme (1995). The definitions for the interfaces can
be found in Appendix A.

4.2 AVL Tree Implementation
There are two primary differences between our implementation and the standard lib-
rary. First, the key-value type was simplified by using a tuple of the key and value,
Key×Value, instead of the dependent key-value pairs favoured by the The Agda Com-
munity. Secondly, the ordering proofs stored in the leaves of the tree were erased
using the @erased notation. This was done in the hopes of improving the runtime
performance of the different map operations since the erasure would remove the rel-
atively slow operations on the ordering proofs such as mklim (the implementation of
which can be found in Appendix A). An evaluation of the performance implications
of this can be found in Section 5.3. It was regrettably impossible to erase the height
or balancing proofs as a number of the map operations case splits on the height and

balancing proofs, making type erasure impossible in the current version of Agda.
These modifications resulted in the map datatype in Listing 4.1

Listing 4.1: Agda definition of AVLMapIndexed.

data AVLMapIndexed (@0 V : Set v) (@0 l u : Key+) : @0 ℕ → Set (k ⊔ v ⊔ ℓ1) where
leaf : {{@erased l<u : l <+ u}} → AVLMapIndexed V l u 0
node : ∀ {hl hr h}

→ ((k , v) : Key × V )
→ (lm : AVLMapIndexed V l [ k ] hl)
→ (rm : AVLMapIndexed V [ k ] u hr)
→ (bal : hl ~ hr ⊔ h)
→ AVLMapIndexed V l u (suc h)

The map operations insertWith, lookup, and delete, as well as the required helper
functions, were directly lifted from the standard library implementation (The Agda
Community, 2023) and modified to erase unnecessary proofs. The implementation
of these are visible in Section A.5. The unionWith function on the other hand, was
based on the algorithm introduced in Adams (1993) and will be discussed in more
detail in Section 4.4.

Listing 4.2: Standard library definition of Any. Reproduced from The Agda Com-
munity, 2023

data Any {V : Value v} (P : Pred (K& V ) p) {l u}
: ∀ {n} → Tree V l u n → Set (p ⊔ a ⊔ v ⊔ ℓ2) where

here : ∀ {h𝑙 h𝑟 h} {kv : K& V} → P kv →
{lk : Tree V l [ kv .key ] h𝑙}
{ku : Tree V [ kv .key ] u h𝑟}
{bal : h𝑙 ∼ h𝑟 ⊔ h} →
Any P (node kv lk ku bal)

left : ∀ {h𝑙 h𝑟 h} {kv : K& V}
{lk : Tree V l [ kv .key ] h𝑙} →
Any P lk →
{ku : Tree V [ kv .key ] u h𝑟}
{bal : h𝑙 ∼ h𝑟 ⊔ h} →
Any P (node kv lk ku bal)

right : ∀ {h𝑙 h𝑟 h} {kv : K& V}
{lk : Tree V l [ kv .key ] h𝑙}
{ku : Tree V [ kv .key ] u h𝑟} →
Any P ku →
{bal : h𝑙 ∼ h𝑟 ⊔ h} →
Any P (node kv lk ku bal)

4.2.1 Any
The Any datatype is traditionally used to express that some predicate 𝑃 holds for
at least one value in a collection. In our case, it will be used to express that some
key and value pair in the map is equal to the key and value pair stipulated in the
predicate. In other words, we will define 𝑘 ↦ 𝑣 ∈ 𝑚 as Any (𝜆(𝑘′ , 𝑣′) → 𝑘 ≡
𝑘′ ∧ 𝑣 ≡ 𝑣′) 𝑚.

Our Any implementation was heavily inspired by the Agda Standard library (The
Agda Community, 2023), Figure 4.2, with three notable differences. Firstly, erasure
in the map datatype requires the erasure of the corresponding proofs in the Any
datatype. Secondly, and more interestingly, we added two additional ordering proofs
to the constructors left and right. This was done to simplify certain proofs where a
comparison of two values and an Any proof might disagree on the order of elements.
An example of such a proof is illustrated in Listing 4.3. The last two cases in
the comparison where the membership proof and the comparison disagree on the
ordering 𝑘 and 𝑘′ are impossible to resolve without the ordering proof 𝑘 < 𝑘′ since
nothing is known about the values prior to the comparison.

Listing 4.3: An example proof using compare.

example : ∀ k {v l u h} (m : AVLMapIndexed V l u h)
→ k ↦ v ∈ m
→ lookup m k ≡ just v

example k (node (k’ , v) lm rm b) (left {{ k<k’ }} prf ) with compare k k’
... | le k<k’ = example k lm prf -- compare agrees with the fact that k < k'
... | eq refl = {!!} -- compare belives that k ≡ k'
... | ge k>k’ = {!!} -- compare believes that k > k' but we know that k < k'

The second modification required the predicate to only operate on the value, as the
Key in the predicate needs to be fixed to a certain value for the ordering proofs to
be definable. This can be thought of as a predicate on keys and values of the type
𝜆(𝑘, 𝑣) → 𝑘 ≡ 𝑘′ × 𝑃 𝑣 where 𝑃 is some predicate on values and 𝑘′ is some key.
Lastly, we erased the maps from the Any datatype as they could conceivably slow
down any operations on or using Any. This modification means that, at compile
time, a Any object consists only of a path to the desired element in the map without
any stored information about the map itself. These modifications resulted in the
Any datatype in Listing 4.4.

21



4. Implementation

Listing 4.4: Agda definition of Any

data Any (P : Pred V ℓ𝑝) {l u : Key+} (k𝑝 : Key) :
∀ {@0 h : ℕ} → @0 AVLMapIndexed V l u h → Set (k ⊔ ℓ1 ⊔ v ⊔ ℓ𝑝) where
here : ∀ {h hl hr} {v : V}

{{@erased l<k : l <+ [ k𝑝 ]}} {{@erased k≤u : [ k𝑝 ] <+ u}}
{@0 lm : AVLMapIndexed V l [ k𝑝 ] hl}
{@0 rm : AVLMapIndexed V [ k𝑝 ] u hr}
{@0 bal : hl ~ hr ⊔ h}
→ P v
→ Any P k𝑝 (node (k𝑝 , v) lm rm bal)

left : ∀ {h hl hr} {(k’ , v) : Key × V}
{@0 lm : AVLMapIndexed V l [ k’ ] hl}
{{@0 k<k’ : [ k𝑝 ] <+ [ k’ ]}}
{@0 rm : AVLMapIndexed V [ k’ ] u hr}
{@0 bal : hl ~ hr ⊔ h}
→ Any P k𝑝 lm
→ Any P k𝑝 (node (k’ , v) lm rm bal)

right : ∀ {h hl hr} {(k’ , v) : Key × V}
{@0 lm : AVLMapIndexed V l [ k’ ] hl}
{@0 rm : AVLMapIndexed V [ k’ ] u hr}
{{@0 k’<k : [ k’ ] <+ [ k𝑝 ]}}
→ {@0 bal : hl ~ hr ⊔ h}
→ Any P k𝑝 rm
→ Any P k𝑝 (node (k’ , v) lm rm bal)

Listing 4.5: Standard library type definition of All. Reproduced from The Agda
Community (2023).

data All {V : Value v} (P : Pred (K& V ) p) {l u}
: ∀ {n} → Tree V l u n → Set (p ⊔ a ⊔ v ⊔ ℓ2) where

leaf : {p : l <+ u} → All P (leaf p)
node : ∀ {h𝑙 h𝑟 h} {kv : K& V}

{lk : Tree V l [ kv .key ] h𝑙}
{ku : Tree V [ kv .key ] u h𝑟} →
{bal : h𝑙 ∼ h𝑟 ⊔ h} →
P kv → All P lk → All P ku → All P (node kv lk ku bal)

4.2.2 All
The All data structure is used to denote that a predicate is universally applicable
across all elements of a collection. An illustrative implementation of this data struc-
ture can be seen in Listing 4.5. Conceptually, the All can be seen as a decorator,
augmenting the original map with new information. One common use case for the
All data structure is to use it as a typing environment by using a type-coercion func-
tion as the predicate. It is for this reason that we needed to extend BasicMap with
an All instance as we whished to utilise it to define the environment in a proof-of-
concept simply typed lambda calculus interpreter. The All data type is generalised
by implementing it as a record, CorrectAll in Listing 4.6, extending the BasicMap,
allowing the developer to leverage the correctness of the map operations.

Listing 4.6: The CorrectAll abstraction extending a BasicMap.

record CorrectAll (Map : Set (ℓ ⊔ ℓ’ ⊔ ℓ1)) (bMap : BMap {ℓ1 = ℓ1} Map) :
Set (suc (ℓ ⊔ ℓ’ ⊔ ℓ1 ⊔ ℓ𝑎)) where

field
All : Pred (K × V ) ℓ𝑎 → Map → Set (ℓ ⊔ ℓ’ ⊔ ℓ1 ⊔ ℓ𝑎)
allInsert : {P : Pred (K × V ) ℓ𝑎} {k : K} {v : V} {m : Map}

→ P (k , v) → All P m → All P (insert k v m)
allLookup : {P : Pred (K × V ) ℓ𝑎} {k : K} {v : V} {m : Map}

→ k ↦ v ∈ m → All P m → P (k , v)

The concrete implementation was lifted from the standard library and extended
with an insertion and lookup function. Furthermore, erasure was used to erase the
ordering proofs in the leaves, resulting in the definition in Listing 4.7. The insertion
and lookup functions can be found in Appendix A, and their helper functions in our
GitHub repository.

4.3 Interface Implementation
The removal of height and balancing proofs is required to implement the abstraction
for our AVL map, as the map implementation has the type

Key+ → Key+ → ℕ → Set k ⊔ ℓ′ ⊔ ℓ1.

The abstraction, on the other hand, expects a map of type Set 𝑘 ⊔ ℓ′ ⊔ ℓ1. The
“removal” is accomplished by “wrapping” the map in a data type that hides the
height and ordering, as shown in Listing 4.8.

We are also required to implement a version of Any that hides the ordering as a direct
consequence of the introduction of AVLMap. This Any instance is implemented in
the same way as AVLMap, meaning that it hides unnecessary information, and it will
be referred to as AnyM going forward. By utilising AVLMap and AnyM it is possible

Listing 4.7: Agda definition of All.

data All (P : Pred (Key × V ) ℓ𝑎) {@0 l u : Key+}
: ∀ {@0 h : ℕ} → AVLMapIndexed V l u h → Set (k ⊔ ℓ1 ⊔ v ⊔ ℓ𝑎) where
leaf : {{ @0 l<u : l <+ u }} → All P leaf
node : ∀ {hl hr h}

→ {p : Key × V}
(let (k , v) = p)
{lt : AVLMapIndexed V l [ k ] hl}
{rt : AVLMapIndexed V [ k ] u hr}
{bal : hl ~ hr ⊔ h}
→ P (k , v)
→ All P lt
→ All P rt
→ All P (node (k , v) lt rt bal)

Listing 4.8: The AVLMap definition, hiding the ordering and height proofs.

data AVLMap (V : Set ℓ’) : Set (k ⊔ ℓ’ ⊔ ℓ1) where
map : ∀ {h} → AVLMapIndexed V ⊥+ ⊤+ h → AVLMap V

to instantiate the three interfaces that make up the abstraction. The full definitions
of these interfaces, as well as the proofs, can be found in Appendix A, while the
implementations can be found in our GitHub repository.

4.4 Union
The Agda standard library contains one possible unionWith implementation, shown
in Figure 4.9. This is a naïve implementation defined using foldr that simply inserts
every key-value pair in one of the finite maps into another.

Listing 4.9: Standard library unionWith.

unionWith : (∀ {k} → Val k → Maybe (Wal k) → Wal k) →
-- left → right → result.
Tree V → Tree W → Tree W

unionWith f t1 t2 = foldr (𝜆 {k} v → insertWith k (f v)) t2 t1

The “naïvety” of this implementation can be explained in two parts: first, due to
the lack of optimisations, and second, because it is very difficult to reason about.
One simple and possible optimisation could be pattern matching on the empty map
to immediately return t1 or t2. This would eliminate the case where you insert the

entirety of one map into an empty map. For the second part, the current imple-
mentation is based solely on the Tree abstraction. This abstraction in the standard
library hides the underlying implementation, meaning that it is not possible to utilise
height, ordering, nor the structure itself, as seen in Section 4.3. The hiding makes it
almost impossible to reason about membership properties for this implementation of
union. One demonstrative example of this can be seen in Listing 4.10. The problem
arises from the fact that the type of insertion in the lambda must match t2, which
is impossible since the insertion will alter the height of m.

Listing 4.10: Example showcasing the problem with the fold-based union. At-
tempting to give the argument m in the hole will produce the error:
proj1 (insertWith k (f v) m) ⊕ h2 != h2.

inT1Union : ∀ {l u : Key+} {h1 h2 : ℕ }
→ (k : Key)
→ {v : V}
→ (f : V → Maybe V → V )
→ {{ l<k : l <+ [ k ] }} {{ k<u : [ k ] <+ u }}
→ (t1 : AVLMapIndexed V l u h1)
→ (t2 : AVLMapIndexed V l u h2)
→ k ↦ v ∈ t1
→ k ↦ v ∈

(foldr (𝜆 (k’ , v’) m → proj2 $ insertWith k (f v) {!m!} ) t2 t1)

With this in mind, we have implemented a unionWith function for AVLMapIndexed
that can be used in the abstraction, that contains the optimisations lacking in the
standard library, and allows for reasoning outside of the abstraction.

Algorithm 1 The unionWith algorithm presented by Blelloch et al. (2016).
function unionWith(𝑓 , 𝑡1 , 𝑡2)

if 𝑡1 is ∅ then return 𝑡2
if 𝑡2 is ∅ then return 𝑡1
(𝑙2, (𝑘2, 𝑣2), 𝑟2) ← 𝑡2
(𝑙1, 𝑣1, 𝑟1) ← splitAt(𝑘2, 𝑡1)
𝑡𝐿 ← unionWith(𝑓, 𝑙1, 𝑙2)
𝑡𝑅 ← unionWith(𝑓, 𝑟1, 𝑟2)
return join(𝑡𝐿, (𝑘2, 𝑓 𝑣1 𝑣2), 𝑡𝑅)

An efficient divide-and-conquer algorithm for unionWith is presented in Blelloch et
al. (2016), and is shown in Algorithm 1. The original algorithm was proposed by
Adams (1993) for weight-balanced trees but formalised for AVL trees by Blelloch
et al.

A join-based unionWith is also potentially more efficient than a fold-based approach.
The complexity of a fold-based unionWith is simple, for 𝑡1 and 𝑡2 with sizes 𝑛 and

𝑚 respectively, a fold-based unionWith must insert all 𝑛 elements into a tree of
size 𝑚. The complexity of insertion is 𝑂(log𝑚) (Knuth, 1998), and as 𝑛 elements
are inserted, the complexity is simply 𝑂(𝑛 log𝑚). The time complexity of a join-
based implementation first requires knowledge of the two helping functions, join
and splitAt. The proof for the complexity of the join-based algorithm is a bit
more complex, so we leave the proof to Blelloch et al. (2016). They show that the
complexity is 𝑂(𝑛 log( 𝑛

𝑚 + 1)) for several join-based functions, unionWith among
them.

Listing 4.11: Type declarations of splitAt and unionWith.

splitAt : ∀ {@0 l u} {h : ℕ}
→ (k : Key)
→ {{@erased l<k : l <+ [ k ]}} → {{@erased k<u : [ k ] <+ u}}
→ (m : AVLMapIndexed V l u h)
→ Split k l u h

unionWith : {h1 h2 : ℕ} → {@0 l u : Key+}
→ (V → Maybe V → V )
→ (t1 : AVLMapIndexed V l u h1)
→ (t2 : AVLMapIndexed V l u h2)
→ UnionReturn t1 t2

We have implemented a join-based unionWith, the full implementation of which is
viewable in our GitHub repository. The type declarations of unionWith and splitAt
can be seen in Figure 4.11. The unionWith function is implemented to operate
exactly according to the algorithm, but with a few caveats due to the data type of
our tree. Our tree is correct-by-construction by ordering and height, meaning that
all trees produced by any functions must maintain ordering and height properties.

The problem of ordering can be solved easily, any order instance can be extended by
an element that is smaller, or larger, than the current limits. This fact can be used
when working with unionWith; if the trees do not have the same ordering already,
the minimum of the two trees can be set as minimum for both, and the maximum
can be extended in the same way.

Height is more problematic. In functions that operate on a single tree, height is often
not an issue, as there is only one height to work with. For example, in insert, a single
element is inserted into a tree of height ℎ, and the tree can achieve a maximum height
of 1+ℎ and a minimum of ℎ. The unionWith function, that operates on two arbitrary
trees require other height restrictions. The arguments of unionWith itself is as one
would expect, it takes a function of type 𝑉 𝑎𝑙𝑢𝑒 → 𝑉 𝑎𝑙𝑢𝑒 → 𝑉 𝑎𝑙𝑢𝑒 and two trees, 𝑡1
and 𝑡2, with arbitrary heights. The algorithm depends on two additional functions,
splitAt and join, that have height constraints visible in the type declaration of gJoin,
Listing 4.13, and the return type of splitAt, Listing 4.12. The problem arises from
splitAt which returns two trees with the constraints that they can achieve at most
the same height as the input tree. With no lower bound on the trees returned from

Listing 4.12: Return types of splitAt and unionWith.

record Split (x : Key) (@0 l u : Key+) (h : ℕ) : Set (k ⊔ v ⊔ ℓ1) where
constructor split
field

value : Maybe V
leftH : ℕ
@0 leftP : leftH ≤ h
leftT : AVLMapIndexed V l [ x ] leftH
rightH : ℕ
@0 rightP : rightH ≤ h
rightT : AVLMapIndexed V [ x ] u rightH

record UnionReturn {@0 l u : Key+} {h1 h2 : ℕ}
(@0 t1 : AVLMapIndexed V l u h1) (@0 t2 : AVLMapIndexed V l u h2)
: Set (k ⊔ v ⊔ ℓ1) where
constructor retval
field

hof : ℕ
tree : AVLMapIndexed V l u hof
@0 prf : hof ≤ (h1 + h2)

Listing 4.13: The type declaration of general join, gJoin. The name join is reserved
for another joining function in the code base. The implementation of gJoin can be
found in the git repository for the project.

gJoin : {hl hr : ℕ} {@0 l u : Key+}
→ ((k , v) : Key × V )
→ AVLMapIndexed V l [ k ] hl
→ AVLMapIndexed V [ k ] u hr
→ ∃ 𝜆 i → AVLMapIndexed V l u (i ⊕ max hl hr)

splitAt it is impossible to prove any lower bound in unionWith as it uses both trees
returned from split.

The upper bound of unionWith is less than or equal to height(𝑡1) + height(𝑡2).
This can be proven by induction (Blelloch et al., 2016). We believe that the bound
is smaller due to the height properties of gJoin. However, we have not been able to
prove this. The bound is explored further in Section 6.2.

Result

With an implementation in place, we will in this chapter present the results of
proving the laws of our theory, as well as the performance evaluation of the imple-
mentation. In addition to proofs and evaluation, we have created a small proof of
concept program used to display the usability and flexibility of our finite map.

5.1 Proofs
We have proven 29 of the 35 correctness proofs for the implementation of the AVL
map abstraction. More specifically, all the insertion, deletion, lookup, and relational
proofs, as well as the union proofs corresponding to Laws 3.24 and 3.25 could be
proven. Listing 5.1 displays one of these proofs, which is required by DeletableMap.
The definition of delete can be seen in Section A.5 in Appendix A. The strategy of
using a set of common helper functions employed by delete-safe’ is used in most of
the proofs to exploit reoccurring parts of the proof structure.

This leaves 6 of the more complex union proofs as well as the insertion-based induc-
tion principle. The union proofs could not be proven as proving the height bounds
was of focus initially, delaying any practical implementation. The induction prin-
ciple, on the other hand, could not be proven as we are unable to show that a finite
number of insertions will reconstruct any map, this is discussed in greater detail and
two possible approaches are presented in Section 6.2.

5.2 Proof of Concept: Lambda Calculus Interpreter
We implemented two versions of a simply typed lambda calculus interpreter, using
the grammar and typing rules shown in Appendix B, Figures B.1 and B.2, to evaluate
the usability of the map abstraction. The first version of the interpreter uses our map
and All abstractions for the typing context and environment, the full implementation
of which can be seen in Appendix B. The second version, on the other hand, uses
the standard library AVL map and All in a nearly identical fashion.

When comparing the ease of use between the two implementations, we found them
almost identical. The only real difference was the fact that the standard library did
not offer insertion and lookup functions for their All data structure. This would

Listing 5.1: Proof for Law 3.22 with helper function types.

in𝑅-join𝑅− : ∀ {l u : Key+} {hl hr h : ℕ}
{k : Key}
{v : V}
{p : Key × V}
(lt : AVLMapIndexed V l [ proj1 p ] hl)
(rt− : ∃ (𝜆 i → AVLMapIndexed V [ proj1 p ] u pred[ i ⊕ hr ]))
(bal : hl ~ hr ⊔ h)
→ @erased [ proj1 p ] <+ [ k ]
→ k ↦ v ∈ (proj2 (join𝑟− p lt rt− bal))
→ k ↦ v ∈ (proj2 rt−)

in𝐿-join𝐿− : ∀ {l u : Key+} {hl hr h : ℕ}
{k : Key}
{v : V}
{p : Key × V}
(lt− : ∃ (𝜆 i → AVLMapIndexed V l [ proj1 p ] pred[ i ⊕ hl ]))
(rt : AVLMapIndexed V [ proj1 p ] u hr)
(bal : hl ~ hr ⊔ h)
→ @erased [ k ] <+ [ proj1 p ]
→ k ↦ v ∈ (proj2 (join𝑙− p lt− rt bal))
→ k ↦ v ∈ (proj2 lt−)

notInJoin : ∀ {l u : Key+} {hl hr h : ℕ}
→ (k : Key)
→ {v : V}
→ (lm : AVLMapIndexed V l [ k ] hl)
→ (rm : AVLMapIndexed V [ k ] u hr)
→ (bal : hl ~ hr ⊔ h)
→ ¬ (k ↦ v ∈ proj2 (join lm rm bal))

del-safe’ : ∀ {l u : Key+} {h : ℕ} (k : Key) {v : V} (m : AVLMapIndexed V l u h)
{{ @erased l<k : l <+ [ k ] }} {{ @erased k<u : [ k ] <+ u }}
→ ¬ (k ↦ v ∈ proj2 (delete k m))

del-safe’ k leaf ()
del-safe’ k (node p lm rm bal) prf with compare k (proj1 p)
... | tri< k<p _ _ = let

prf’ = in𝐿-join𝐿− (delete k {{ p≤u = [ k<p ]𝑅 }} lm) rm bal [ k<p ]𝑅 prf
in del-safe’ k lm {{ k<u = [ k<p ]𝑅 }} prf’

... | tri≈ _ refl _ = notInJoin k lm rm bal prf

... | tri> _ _ p<k = let
prf’ = in𝑅-join𝑅− lm (delete k {{ [ p<k ]𝑅 }} rm) bal [ p<k ]𝑅 prf

in del-safe’ k rm {{ [ p<k ]𝑅 }} prf’

require more work from an end-user perspective if we were to treat the two maps as
two different libraries.

However, it should be mentioned that our interpreter can use any implementation
of BasicMap for its typing context and environment. This results in a great degree
of flexibility since the interpreter itself does not depend on any specific underlying
implementation, allowing a developer to change the map implementation to a more
efficient or simpler version without having to rewrite any of their code.

5.3 Performance Comparison
The performance of our AVL map implementation was evaluated by compiling it
into Haskell using Agda’s GHC backend (Benke, 2007) and then benchmarking it
against the standard library version using Criterion (O’Sullivan, 2014/2024). The
benchmarks were performed using an 8-core AMD Ryzen 7 7700 processor with 32
GB of DDR5 RAM running at 6000 MHz. The map implementations were com-
pared in terms of execution time for insertion and union operations. The insertion
operations were compared by measuring the execution speed when inserting 1000,
10 000, 100 000, 1 000 000, 10 000 000 elements in ascending order into an empty
map, the results of which can be seen in Figure 5.1.

Figure 5.1: Elapsed time for inserting values in the range 0..𝑛 into an empty map,
with logarithmic 𝑥 and 𝑦 axes.

Union, on the other hand, was compared in two different ways. First, by merging
two overlapping maps where one of the maps contains 100 000 elements and the
other contains 1000, 10 000, or 100 000 elements. The results of this can be seen in

Figure 5.2. Second, by merging an empty map with a map containing 1000, 10,000,
or 100,0000 elements. The results of which are summarised in Table 5.1 and fully
presented in Figure C.2. Each of these comparisons was performed using the same
set of key-value pairs to ensure fairness.

(a) 𝑛 is fixed with 100 000 elements.

(b) 𝑚 is fixed with 100 000 elements.

Figure 5.2: Times for merging two overlapping maps, 𝑛 and 𝑚, using union. One
of the maps contains 100,000 elements while the other one contains 1000, 10,000, or
100,000 elements.

Operation Our implementation Standard library
union ∅ ∅ 13.4 ns 7.27 ns
union 𝑛 ∅ 13.11 ns 6.505 ns
union ∅ 𝑛 13.28 ns 22.83 ms

Table 5.1: Times for merging an empty map with a map, 𝑛, containing 100,000
elements.

We can conclude based on the results in Figure 5.1 and 5.2 as well as those in
Table 5.1 that our AVL map implementation performs worse in all non-empty cases
with the exception of performing union when the left tree is disproportionally larger
than the right tree, which can be seen in Figure 5.2.

5.4 Memory Profiling
In addition to performance testing, we have performed basic memory profiling. GHC
has great support for profiling, but as we are first compiling the Agda code to Haskell,
and from Haskell to an executable, we were regretfully not able to model the profiling
as precisely as we wished. However, it was possible to compile the program with the
-rtsopts flag, enabling us to perform rudimentary profiling using the GHC run-time
options. Although these results are not as accurate as we desired, we still believe
that the results are interesting and relevant.

To force evaluation and consistent results, we used the same programs for which per-
formance evaluation was performed, with the added change that each benchmark

Figure 5.3: Maximum residency for inserting values in the range 0..𝑛 into an empty
map, with logarithmic 𝑥 and 𝑦 axes.

was performed separately to produce relevant memory usage statistics. Both max-
imum residency, visible in Figure 5.3, and total memory usage for insert indicates
that erasure does not have a material impact on memory usage. On the other hand,
the maximum residency for union, visible in Figures 5.4 and 5.5, indicates that the
chosen algorithm has a greater impact on memory usage than erasure. Graphs show-
ing the total memory usage for these benchmarks as well as the linear version of the
insertion benchmark, can be found in Appendix D.

Figure 5.4: Maximum residency for for union where 𝑛 is fixed with 100 000 ele-
ments.

Figure 5.5: Maximum residency for for union where 𝑚 is fixed with 100 000
elements.

Discussion

In this chapter, the extended theory, our implementation, and the results of perform-
ance and memory evaluation will be discussed. In addition, we will discuss correct-
by-construction for finite maps and union properties of AVL and size-balanced trees.

6.1 Correct-by-Construction Key-Value Properties
When implementing a finite map, there are aspects that are well suited to be correct
by construction as described by the introduction of this thesis. The implementation
described in Chapter 4 implement these correct-by-construction aspects. An initial
goal of this thesis was to explore the introduction of more correct-by-construction
properties in finite map implementations. This could, for example, be to introduce
key-value membership properties within the data structure of a tree. In the initial
phase of research we entertained this thought but quickly realised that it is rather
difficult to achieve. The primary reason for not integrating this into our implement-
ation is two-fold:

• Proofs generated for a single tree does not necessarily hold for any other, i.e., a
proof of membership for a specific element in one tree is potentially worthless
upon insert, deletion, or any other operation that modifies the tree in some
way.

• If correctness proofs are bundled with the collection, they all need to be con-
firmed relevant and updated on every modification that can potentially change
elements in some way, as per the first point.

With that in mind, it was deemed that correct-by-construction key-value properties
were not maintainable for this thesis. Membership correctness has instead been
proven for most operations in our implementation.

A perhaps obvious approach that has not been explored more than theoretically
by us is to keep a proof of membership in some strictly ordered data structure or
accumulator. An accumulator would require a hash function producing hashes of
keys that are all relatively prime so that dividing would give appropriate answers.
The accumulator approach is a fun idea but not exactly feasible due to the difficulties
of proving and defining the hash function. As for ordered structures, the obvious
answer would be to use a set. Good set representations are, for example, tree
structures because of their fast lookup, insertion, and deletion. This idea never left

the theoretical stage for good reason, as it requires maintaining two trees of the
same size and with almost the same content. At that point, one might as well use
the original structure to produce the proofs.

6.2 Union and Join
The union implementation of this thesis is lacking in one aspect, the height bound.
The height bound is practically irrelevant at run-time, but it is interesting for reason-
ing with since it expresses a sort of “size” of AVL trees. Because of this relationship
one wants to assert that the heights do not increase, or decrease, more than is
possible. A more exact bound obviously expresses this more exactly. Due to the
characteristics of the helper functions, it is not possible to provide a lower bound,
and the upper bound is extraordinarily high at height(𝑡1) + height(𝑡2)!
We believe that the actual bound is lower. To reason why we first need to know
the bounds of the helper functions, for gJoin we know that the bound is narrow,
with 𝑖 + max(height(𝑡1),height(𝑡2)) where 𝑖 ∈ {0, 1}. splitAt, on the other hand,
works by traversing a tree 𝑡 to the point where the specified element, hereby the
“pivot”, should be located in the tree, and returns two trees, one consisting of all
elements in 𝑡 less than the pivot and the other consisting of all elements in 𝑡 larger
than the pivot. This gives us two cases; either the pivot can exist within 𝑡, or it
is outside the valid range of elements. In the outside case, both left and right, 𝑡
itself will be returned, resulting in one tree with height height(𝑡) and the other
with height 0. If the pivot exists inside 𝑡 then a maximum height of height(𝑡) is
possible for either tree. The combined minimum bound is obviously 0.

A height of interest when implementing split is the “minimum upper bound”. The
minimum upper bound is defined by us to be part of the bound of the largest returned
tree. This is interesting for the implementation as it gives the developer more to
work with. The complete upper bound of the trees returned from split could then
be defined as:

height(𝑡) − 2 ≤ max(height(𝑡𝑙),height(𝑡𝑟)) ≤ height(𝑡).

Where 𝑡𝑙 and 𝑡𝑟 are the output trees of splitAt. For example, splitting the following
tree at 4:

3

2

1

5

4 6

7

The result of the split is the value associated with 4 and the two trees:

2

1 3

6

5 7

This bound is also supported by the join bound, since the largest tree input to join
in splitAt is the lowest of the two children of 𝑡.
When reasoning about the bound of unionWith we state the condition that one tree
is smaller than or equal to the other. Let unionWith be executed with the three
arguments 𝑓 , 𝑡1 with height ℎ1, and 𝑡2 with height ℎ2. Here we state that ℎ1 ≤ ℎ2,
i.e., 𝑡1 is smaller than or equal to 𝑡2.

We propose that the lower bound is max(ℎ1, ℎ2). The reasoning for this is that since
𝑡2 is divided at its root node, 𝑘, producing two balanced trees that are recursively
unioned with the results of splitAt 𝑘 𝑡1 it is not possible for the height to decrease.
This is because any addition to the heights from 𝑡2 and its children can not decrease
in height according to the height bound of join, and as the two trees are balanced
at the pivot the height will always increase to the original height of the tree. This
lower bound remains in the case where ℎ1 ≡ ℎ2 by the same reasoning.

For the upper bound, we propose a limit of 1 + max(ℎ1, ℎ2). We have not been
able to prove this, but we strongly believe that it is correct due to the properties of
splitAt and join, and the fact that no counter-example has been found. Regardless
of whether our proposed bound is correct or not, it is certain that the proven bound
of max(height(𝑡1),height(𝑡2)) is far above what is reasonable for union of trees
higher than 0 and 1.

6.2.1 The Benefits of Join
Blelloch et al. (2016) demonstrates the extreme flexibility of the join function by
using it to define insert, delete, union, intersect, and difference. We provide a general
join, meaning that it allows one to join any two AVL trees of varying heights. Our
usage of the join function is limited to the union functionality, which means that
it is used in splitAt and unionWith. However, similar to Blelloch et al. (2016), this
generalised join function could be used to define the complete set of operations in
further work on provably correct collections. This would greatly simplify the proof-
writing process since a small number of helper functions could be used in a large
number of different proofs. An example of such a function would be a function that
states that if 𝑘 ↦ 𝑣 ∈ 𝑡1 then 𝑘 ↦ 𝑣 ∈ gJoin 𝑡1 𝑡2. Such a function would be useful
for proving a number of different insertion, deletion, and union properties.

6.2.2 Union in Size-Indexed Trees
Another approach that has not been fully explored is the possibility of using some
size-indexed tree, e.g., weight-balanced (Adams, 1993), instead of AVL trees. The
size of a tree is simply the number of elements, meaning size(leaf) ≡ 0 and

size(node _ l r) ≡ 1 + size(l) + size(r). Weight-balanced trees are balanced by
some factor or ration of the size of the tree. Trees indexed by size can be beneficial
for union as the invariant for join and splitAt are more exact.

For example, a comparison of the height and size types of join, 𝑖 + max(ℎ1, ℎ2),
for 𝑖 ∈ {0, 1}, and 1 + 𝑠2 + 𝑠2, shows that the height relation is less specific. It
is not possible to assert the exact height at a glance, whereas the size will always
increase by exactly 1 if the value added is a pivot between the two trees, which is a
requirement by the function. Similarly, for splitAt the size must be 𝑖 + 𝑠 ≡ 𝑠𝑙 + 𝑠𝑟,
for 𝑖 ∈ {0, 1}, given an input tree of size 1 + 𝑠 and output trees with sizes 𝑠𝑙 and
𝑠𝑟. This can be made even more exact if splitAt is divided into two functions, one
for the case where the desired element exists in the map and another for the case
where it does not. The unionWith bound for size-indexed trees is exactly:

max(size(𝑡1), size(𝑡2)) ≤ size(𝑡𝑜𝑢𝑡) ≤ size(𝑡𝑙) + size(𝑡2).

The bounds for unionWith and splitAt are proven for size-indexed trees in the file
SizeTree.agda in our GitHub repository, but join is postulated since its size invariant
is obviously correct.

As shown, it is simpler to reason with splitAt, join, and unionWith using size-indexed
trees, but the work needed to implement other basic operations has not been ex-
plored. One possible approach would be to convert AVL trees to weight-balanced
trees when working with union, but such a conversion requires counting all elements,
which could potentially make it inefficient.

Listing 6.1: A recurrence describing the minimum number of nodes in an AVL-
tree (Adelson-Velskii & Landis, 1962)

minSize : ℕ → ℕ
minSize 0 = 0
minSize 1 = 1
minSize (suc (suc h)) = 1 + minSize h + minSize (suc h)

In addition to inefficiency, we see that a height-balanced tree using the AVL invariant
can contain a varying number of nodes, the minimum of which can be expressed by
the recurrence in Listing 6.1 and the maximum is described by 2ℎ − 1.

The recurrence describes a tree that adheres to the AVL-invariant, but where the
children of each tree always differ in height by exactly one. For the maximum we
consider that the number of nodes in each layer doubles from the next to the other,
e.g.,

1 + 2 + 4 + 8 + ⋯+ 2ℎ−1 =
ℎ−1
∑
𝑖=0

2𝑖.

As our implementation considers trees of size 1 to have height 1 we must start from
0 and go to ℎ − 1. This geometric series can be expressed as 2ℎ − 1.

The minimum and maximum number of nodes in a balanced AVL tree suggests that
a weight-balanced tree may allow for trees that do not hold according to the AVL-
invariant. This implies that it is only possible in some cases to easily transform a
weight-balanced tree into a height-balanced tree.

6.3 Inductive Principle
Collins and Syme show that the insertion-based induction principle is provable, but
we were unable to prove this for AVL trees. The main issue is the fact that we
are unable to use structural induction, as the definition of the principle is structure
agnostic, which forces us to induce on size of the tree. This causes a number of
issues; the most significant being that we need to be able to remove some element
from a map such that inserting the element into the resulting map produces the
original map. This need arises from the step function with the type:

∀𝑘 𝑣 𝑚2. 𝑃 𝑚2 → 𝑘 ∉ 𝑚2 → 𝑃 (insert(𝑘, 𝑣) 𝑚2).
The step function represents the inductive step and is used when size(𝑚1) >
0. In this case, we want to show that 𝑃 𝑚1 holds, but we can only show that
𝑃 (insert(𝑘, 𝑣) 𝑚2) holds for some 𝑘, 𝑣, and 𝑚2.

We have found two possible ways of accomplishing this, but have not been able to
implement them in Agda, as they rely on functions or proofs that we were unable
to complete. The first approach uses Postulate 1, defining a new function with the
purpose of finding an element that can be reinserted to produce the original tree.

Postulate 1. Assume that there exist some function separate that given a map 𝑚,
containing at least one element, produces ((𝑘 , 𝑣) , 𝑚′) such that insert (𝑘 , 𝑣) 𝑚′ ≡
𝑚 and ∀𝑘 𝑣 𝑚 𝑚′. separate 𝑚 ≡ ((𝑘 , 𝑣) , 𝑚′) → 𝑘 ∉ 𝑚. The induction principle
is provable if separate is definable.

Proof. We want to show that

𝑃 ∅ × (∀𝑘 𝑣 𝑚. 𝑃 𝑚 → 𝑘 ∉ 𝑚 → 𝑃 (insert(𝑘 , 𝑣) 𝑚)) → (∀𝑚 → 𝑃 𝑚)
We will perform induction on the size of the 𝑚.

Base case (size(𝑚) = 0): 𝑃 𝑚 hold since we know that 𝑃 ∅ holds.

Inductive step (size(𝑚) > 0): Assume that 𝑃 𝑚′ holds, where separate 𝑚 ≡
((𝑘 , 𝑣) , 𝑚′). Using the step function, we get 𝑃 (insert (𝑘 , 𝑣) 𝑚′) which is equal
to 𝑃 𝑚 according to the postulate.

The main issue with Postulate 1 is the definition of separate, which needs to cleverly
remove an element so that insertion reverses the removal. Some early testing of this
approach seems to indicate that separate is definable for AVL trees, but more work
is required to fully verify the viability of this approach.

The second approach uses Postulate 2, which uses insertion and deletion to achieve
the same result as Postulate 1.

Postulate 2. We are able to prove the induction principle by assuming that ∀ 𝑘 𝑣 𝑚. 𝑘 ↦
𝑣 ∈ 𝑚 → insert (𝑘 , 𝑣) (delete 𝑘 𝑚) ≡ 𝑚.

Proof. We want to show that

𝑃 ∅ × (∀𝑘 𝑣 𝑚. 𝑃 𝑚 → 𝑘 ∉ 𝑚 → 𝑃 (insert (𝑘 , 𝑣) 𝑚)) → (∀𝑚 → 𝑃 𝑚)

We will perform induction on the size of the 𝑚.
Base case (size(𝑚) = 0): 𝑃 𝑚 holds since we know that 𝑃 ∅ holds.

Inductive step (size(𝑚) > 0): Assume that 𝑃 (delete 𝑘 𝑚), where (𝑘, 𝑣) is the
pivot element in 𝑚, holds. Using Law 3.19, we can show that 𝑘 ∉ delete 𝑘 𝑚. Using
the given step function we get 𝑃 (insert (𝑘, 𝑣) (delete 𝑘 𝑚)), which is equal to
𝑃 𝑚 according to postulate 2.

The second approach is more appealing as it does not define any new function but
instead uses two existing functions that have been extensively verified. The main
issue lies in proving the equivalence between the original map and the modified
one. One might be able to simplify the proof by specifying it for when 𝑝 is deleted
from node 𝑝 𝑙𝑚 𝑟𝑚 𝑏𝑎𝑙. This would simplify the proof, since we would not have to
search the entire tree structure for the key-value pair that should be removed and
reinserted.

6.3.1 Equality and Induction
This section has thus far used _≡_ when reasoning about the equality of maps, which
is not exactly ideal as its definition of equality requires the trees to have identical
structures. The focus on _≡_ stems from the fact that the current formulation of
the induction principle requires us to rewrite insert k v m’ into m in the final step
of the proof.

Another possible approach is to use the _≐_ relation, discussed in Chapter 3, when
reasoning about the equality of the trees. This approach would be beneficial since
we would no longer have to consider the structure of the trees when establishing
that they are equal, which would greatly simplify the implementation. This would
be rather simple to implement as all we would have to do is to require that the
predicate, P, respects the relation _≐_. This would essentially provide us with a
function of the type:

∀ m m’ → m ≐ m’ → P m → P m’.

Such a function could be used to “rewrite” the result of the step function into the
desired type at the end of the computation, given that they fulfil extensional equality
as per _≐_. This approach could allow us to use existing insertion and deletion
proofs when implementing Postulate 2.

However, there are still issues with this approach. First, it would limit the number
of predicates that can be used in the induction principle as all of them will not
necessarily respect _≐_. This would also place a greater burden on the developer
as they would have to prove that their predicate respects _≐_. Second, and more

importantly, the height parameterisation of the current AVL map implementation
makes it rather difficult to prove that a combination of operations, e.g., insert and
delete, is equal to the original map even when using _≐_. This stems from the fact
that we cannot predict whether or not an object will change the height of the map.
This seems to indicate that height parameterisation might not be the best approach
for this kind of functions and that size parameterisation could possible simplify or
outright solve these problems.

6.4 Size and Height Parameterised AVL-Trees
A reoccurring issue for both the induction principle and union is the fact that the
height of a map is not enough to reason about the number of elements. This naturally
raises the question: why not parameterise on the number of elements instead of
the height of the map? This modification could simplify the implementation of
the induction principle. Additionally, this would also simplify the union proofs,
as discussed in Section 6.2.2. However, there are a couple of drawbacks to this
approach.

Firstly, parameterising on both the height and number of elements would not make
much sense since we would be conveying the same information twice. We would
therefore be required to express the balancing of the trees in terms of their size. This
could make the balancing proofs more complicated, as well as any of the operations
that modify the balancing of the map, such as insertion, deletion, and union. This
would additionally complicate the correctness proofs for these operations, since there
would be a greater number of possible cases when rotating the trees, or more precisely
there would be the same number of possible rotations, but a greater number of cases
that result in these rotations.

Secondly, if we instead parameterise on both the height and the size, then we might
have to introduce a connection between the size and height. More specifically, Agda
might not be able to understand the connection between height and size since the
height does not increase at the same rate as the size. This might cause problems
when proving certain properties about the map operations or when defining the
operations themselves.

6.5 Performance Evaluation
The performance results presented in Section 5.3 show that our implementation has a
consistent overhead compared to the standard library implementation. The insertion
results in Figure 5.1 clearly show this, as our insertion function perfectly mirrors the
standard library, with the exception that we erase the ordering proof updates. One
would therefore assume that our insertion function should be at least as fast if not
faster than the standard library version, since we should be performing less work in
each insertion because of the erasure. This seems to indicate that there is something
intrinsic in our implementation that is slowing us down. One possible explanation
could be that the instance argument ordering proofs are somehow slowing down the

execution time; this would be very strange as these proofs are erased and should
therefore have no impact on the performance. Another reason could be the added
abstraction, recall that any call to a tree operation must be performed through the
abstraction.

The union results in Figure 5.2 and in Table 5.1 are more reasonable, since our union
implementation, specifically splitAt, has to produce a large number of different proofs
to be able to prove the correctness of the height of the resulting tree. It is also clearly
visible in the figures that our unionWith is more optimal for some specific inputs,
namely when the first input tree is smaller than the second input tree. Possible
optimisations for this in production can be to first compare the height of the trees
and swap them accordingly when calling unionWith. This optimisation could also
be performed for the standard library.

6.5.1 Effects of Erasure on Performance
The @erasure and @0 notation in Agda signify run-time irrelevance, meaning that
anything marked with that will be ignored during compilation. In our library, we
extensively denote the code with this run-time irrelevance notation. We have not
noticed any significant performance gains from the introduction of erasure. It is not
possible for us to tell whether the consistent overhead of our implementation negates
any potential gain from run-time irrelevance. It could be interesting to compare our
code without erasure but as this requires manually deleting all erasure notations
from the code base we decided against it. In addition to the extra labour late in
the development process the results of the performance comparison show that ours
performs very similarly to the standard library that does not use erasure.

Agda2hs (Cockx et al., 2022) is an alternative to MAlonzo that aims at producing
readable Haskell code. The Agda2hs program is lacking some features necessary
for our code, specifically type-level lambdas and module parameterisation. Worth
noting is that these features are not possible to implement in Haskell. At the time
of performance evaluation, it was deemed out of scope to modify the code base to
be able to compile to Haskell using Agda2hs.

6.5.2 Effects of Erasure on Memory
The memory profiling performed using GHC’s run-time options shows no correlation
between memory usage and erasure. Which is strange as one would expect there to
be a noticeable difference. We believe that this could be caused by Haskell’s call-by-
need evaluation strategy, since Haskell might not need to evaluate any of the values
that are run-time irrelevant in our map but exist in the standard library. If that
is the case then the erasure notation is perhaps unnecessary in most of our cases.
The most useful aspect of the notation would be if there are very inefficient proofs,
which our implementation currently lacks for execution of the operations.

The memory usage results combined with the performance results would indicate
that erasure might not be beneficial in a call-by-need language, at least in terms of
performance or memory usage.

6.5.3 On Correctness and Performance
As is explained and shown the performance of our map implementation is very
similar to the standard library when using the same algorithms. This leads us to
the question: are there other ways that one can improve the performance, and at
what cost of correctness? We believe that there are three main paths one can take
to try to answer this:

1. Simplify the implementation as much as possible such that there are no proper-
ties that need to be maintained in the implementation itself. The correctness
would in this case need to be proven completely outside of the implementation,
much alike the laws described and proven by us.

2. Use an implementation that leverages correct-by-construction such that prop-
erties are correct by design. The difficulty with this case is that correct-by-
construction may require in-line proofs in the implementation, which can slow
it down. In addition to proofs in the implementation, one would also need
to make sure that the types are correct, which may be difficult, as shown for
union in Section 6.2.

3. Have two implementations, one that is easy to reason with, for example point
two in this list, and another that is very efficient, for example point one. Given
that one implementation is easier to reason with one could prove that and then
show an equivalence between the two implementations, and mark the slow as
run-time irrelevant.

As one could perhaps guess, we have been striving towards an implementation like
point two above. There are still improvements that could be made, for example, our
implementation shows a consistent overhead on insertion, and performing union is
in general slower despite theoretically better complexity.

The question remains whether to prioritise correctness or performance. There is
no clear answer to this question unfortunately, as it depends on the purpose of the
program. Our implementation was done purely of theoretical interest, beginning
with exploring how reasonable it was to use correct-by-construction key-value mem-
bership properties and ending with attempting to improve the standard library. We
are generally satisfied with the performance and memory usage of our implement-
ation, despite hoping for better results given improved union implementation and
the amount of code marked as run-time irrelevant.

Real world scenarios of course require different priorities, for example, security con-
cerns or cost analysis of taking the extra time to formally verify your software instead
of developing “good enough” software, that works but is not necessarily verified.

The work presented in this thesis is guided by the following questions, in addition
to providing a provably correct and usable collection for Agda:

1. Which commonly used functions are lacking proofs and is it possible to add
and prove these?

2. How much of an underlying map implementation can be abstracted while
retaining the usability and flexibility of the map?

3. Can the implementation be written such that parts of the code used only for
verification are erased during compilation?

Initially, it was believed that most, if not all, of the functions available for collections
in Agda lacked proofs. This was not the case. However, the laws provided in this
thesis extend the existing proofs and add additional laws for deletion and union that
previously had none. The majority of the laws are proven, the only exception being
the inductive principle and some proofs for union. These exceptions stem from the
AVL implementation since reasoning with the height does not express the correct
properties for the inductive principle, and similarly, for union, the bounds of the
height were not deemed satisfactory, so time was spent on finding, and proving new
bounds.

The answer to the second question is that a lot of underlying details can be abstrac-
ted. This of course comes with consequences too; to allow for the greatest flexibility
of implementation, the inductive principle was defined using the insert operation,
which led to us not being able to prove it correct for our implementation. But in
general, the abstraction is flexible and usable, as seen in the simply typed lambda
calculus interpreter example in Section 5.2.

Finally, we believe that it is possible to erase a lot of information from the structure,
but this was not shown to improve the performance of our implementation. By
analysis of the generated Haskell code, it is unclear what impact run-time irrelevance
has on our code. Due to constraints stemming from parameterised modules and type-
level lambda abstraction, we were not able to use a different program (Cockx et al.,
2022), that is known to completely remove parts marked as run-time irrelevant, to
generate readable Haskell.

To conclude, we have successfully implemented a collection in Agda that is provably
correct, and fully proven for all operations except unionWith.

7.1 Future Work
Future work is presented in Chapter 6, and is briefly summarised here. As we only
add proofs for two new operations, there is more work available for map operations,
namely implementations suitable for proving and reasoning with as well as providing
laws that express appropriate behaviour.

The height bound of union in AVL trees is something we have attempted to define,
but this has been unsuccessful. We believe that the bound is tighter than suggested
by Blelloch et al., but we were unable to show it. The believed upper bound is
1 + max(height(𝑡1),height(𝑡2)) for the input trees 𝑡1 and 𝑡2, and while we were
unable to prove it, we have not been able to find counterexamples. We hope that
future work will either prove or disprove our claim and provide an appropriate bound.

As described in Chapter 6, weight-balanced trees make it easier to reason with union
properties. An implementation using weight-balanced trees that are proven correct
is something we have not explored but is a reasonable alternative considering the
struggles of union described in this report. An implementation that uses size for
reasoning could additionally be beneficial for proving the inductive principle.

Correct-by-construction aspects of collections can also be further explored, for ex-
ample, by bundling membership proofs with operations that insert or delete elements.
Our implementation contains proofs that operate similarly, but are not as tightly
integrated as it could be. Other areas of future research could be how the proofs
are constructed, e.g., other set representations. Future work on proof construction
and usage can show what is efficient, most maintainable, or most appropriate for
reasoning.

The use of Agda2hs (Cockx et al., 2022) instead of the MAlonzo compiler (Benke,
2007) is something we would have liked to have done, but the implementation is
currently dependent on features not supported by Agda2hs. Future developers could
accommodate the implementation or write their own, to test the impact of run-time
irrelevance and to obtain Haskell code that is easier to analyse.

A
Definitions of Abstractions, an
Ordering Function, and Some

Function Implementations

This appendix contains the definitions of the following abstractions and functions:
records BMap, DMap, and MMap, functions mklim, insertWith, insert, delete, and
lookup.

A.1 Definition of the BMap Record

module _ {ℓ1 : Level} {K : Set ℓ} {V : Set ℓ’} where
record BMap (Map : Set (ℓ ⊔ ℓ’ ⊔ ℓ1)) : Set (lsuc (ℓ ⊔ ℓ’ ⊔ ℓ1)) where

constructor mkMap
field
∅ : Map
singleton : K → V → Map
_∈_ : K → Map → Set (ℓ ⊔ ℓ’ ⊔ ℓ1)
_↦_∈_ : K → V → Map → Set (ℓ ⊔ ℓ’ ⊔ ℓ1)
lookup : Map → K → Maybe V
lookup∈ : ∀ {k m} → k ∈ m → V
insertWith : K → (Maybe V → V ) → Map → Map

insert : K → V → Map → Map
insert k v m = insertWith k (𝜆 _ → v) m

_↦_∉_ : K → V → Map → Set (ℓ ⊔ ℓ’ ⊔ ℓ1)
k ↦ v ∉ m = ¬ (k ↦ v ∈ m)

_∉_ : K → Map → Set (ℓ ⊔ ℓ’ ⊔ ℓ1)
k ∉ m = ¬ (k ∈ m)

_⊆_ : Map → Map → Set (ℓ ⊔ ℓ’ ⊔ ℓ1)
n ⊆ m = ∀ k v → k ↦ v ∈ n → k ↦ v ∈ m

_≐_ : Map → Map → Set (ℓ ⊔ ℓ’ ⊔ ℓ1)
n ≐ m = (n ⊆ m) × (m ⊆ n)

field
refl⊆ : Reflexive _⊆_
trans⊆ : Transitive _⊆_
refl≐ : Reflexive _≐_
sym≐ : Symmetric _≐_
trans≐ : Transitive _≐_

↦∈To∈ : ∀ {k v m} → k ↦ v ∈ m → k ∈ m
∈To∈↦ : ∀ {k m} → k ∈ m → ∃ (𝜆 v → k ↦ v ∈ m)

∈↦-∅ : ∀ k v → k ↦ v ∉ ∅
∈-∅ : ∀ k → k ∉ ∅

∈-singleton : ∀ k k’ v v’ → k ↦ v ∈ singleton k’ v’ → k ≡ k’ × v ≡ v’

ips : (P : Map → Set (ℓ ⊔ ℓ’))
→ P ∅ × (∀ m → P m → ∀ k v → k ∉ m

→ P (insertWith k (𝜆 _ → v) m))
→ (∀ m → P m)

---------------------------------------------------------------------------------
-- Insertion and lookup properties
---------------------------------------------------------------------------------
lookup-∅ : ∀ k → lookup ∅ k ≡ nothing

∈⟹lookup : ∀ m k {v} → lookup m k ≡ just v → k ↦ v ∈ m
lookup⟹∈ : ∀ m k v → k ↦ v ∈ m → lookup m k ≡ just v

∉⟹nothing : ∀ m k → k ∉ m → lookup m k ≡ nothing
nothing⟹∉ : ∀ m k → lookup m k ≡ nothing → k ∉ m

lookup≡lookup∈ : ∀ k m → (k∈m : k ∈ m) → just (lookup∈ k∈m) ≡ lookup m k

mapsTo : ∀ {m k} → (k∈m : k ∈ m) → k ↦ lookup∈ k∈m ∈ m

lookup-insert : ∀ k m f
→ lookup (insertWith k f m) k ≡ just (f (lookup m k))

ins-comm : ∀ k k’ v v’ m
→ k ≢ k’
→ insert k v (insert k’ v’ m)

≐ insert k’ v’ (insert k v m)

∈-ins : ∀ m k x v f
→ x ↦ v ∈ (insertWith k f m)
→ (x ≡ k) ⊎ x ↦ v ∈ m

insert∈ : ∀ k v m → k ↦ v ∈ (insert k v m)

∈insert : ∀ k {v} {v’} m → k ↦ v ∈ (insert k v’ m) → v ≡ v’

insert-safe : ∀ {k k’ v v’ m}
→ k ↦ v ∈ m → k ≢ k’ → k ↦ v ∈ (insert k’ v’ m)

ip : (P : Map → Set (ℓ ⊔ ℓ’))
→ P ∅ × (∀ m → P m → ∀ k v → P (insertWith k (𝜆 _ → v) m))
→ (∀ m → P m)

ip P (b , s) mp = ips P (b , 𝜆 m x k v _ → s m x k v ) mp

A.2 Definition of the DMap Record

record DMap (Map : Set (ℓ ⊔ ℓ’ ⊔ ℓ1)) : Set (lsuc (ℓ ⊔ ℓ’ ⊔ ℓ1)) where
constructor mkDMap

field
bMap : BMap Map
delete : K → Map → Map

---------------------------------------------------------------------------------
-- Deletion properties
---------------------------------------------------------------------------------
del-∉ : ∀ {k m} → k ∉ m → delete k m ≐ m
del-∈ : ∀ {k m} → k ∉ delete k m
del-safe : ∀ {k k’ v m} → k’ ↦ v ∈ m → k ≢ k’ → k’ ↦ v ∈ delete k m
del-noAdd : ∀ {k k’ v m} → k ↦ v ∈ delete k’ m → k ↦ v ∈ m
del-removeK : ∀ {k v m} → k ↦ v ∉ delete k m
del-comm : ∀ k k’ m → delete k (delete k’ m) ≐ delete k’ (delete k m)

A.3 Definition of the MMap record

record MMap (Map : Set (ℓ ⊔ ℓ’ ⊔ ℓ1)) : Set (lsuc (ℓ ⊔ ℓ’ ⊔ ℓ1)) where
constructor mkMMap

field
bMap : BMap Map
unionWith : (V → Maybe V → V ) → Map → Map → Map

---------------------------------------------------------------------------------
-- Union Properties

--------------------------------------------------------------------------------
∪-∅𝐿 : ∀ m f → unionWith f m ∅ ≐ m
∪-∅𝑅 : ∀ m f → unionWith f ∅ m ≐ m
∪-∅ : ∀ m f → unionWith f m ∅ ≐ unionWith f ∅ m
∪-∈ : ∀ m1 m2 f k

→ k ∈ unionWith f m1 m2
→ k ∈ m1 ⊎ k ∈ m2 ⊎ (k ∈ m1 × k ∈ m2)

∪-∈’ : ∀ m1 m2 f k
→ k ∈ m1 ⊎ k ∈ m2
→ k ∈ unionWith f m1 m2

∪-safe : ∀ k v1 v2 m1 m2 f
→ k ↦ v1 ∈ m1
→ k ↦ v2 ∈ m2
→ k ↦ f v1 (just v2) ∈ unionWith f m1 m2

∪-safe-left : ∀ k v m1 m2 f
→ k ↦ v ∈ m1
→ k ∉ m2
→ k ↦ f v nothing ∈ unionWith f m1 m2

∪-safe-right : ∀ k v m1 m2 f
→ k ∉ m1
→ k ↦ v ∈ m2
→ k ↦ v ∈ unionWith f m1 m2

A.4 The Implementation of the mklim Function

@erased mklim : ∀ {l u h}
→ AVLMapIndexed V l u h
→ l <+ u

mklim (leaf {{p}}) = p
mklim {l} {u} (node p lt rt bal) = trans+ l (mklim lt) (mklim rt)

A.5 Implementation of insertWith, insert, delete,
and lookup

The left and right join functions join a key-vale pair with two sub-trees, performing
a single or double rotation if necessary.

insertWith : {@0 l u : Key+} {@0 h : ℕ} (k : Key) (f : Maybe V → V )
{{@erased l≤u : l <+ [ k ]}} {{@erased p≤u : [ k ] <+ u}}
→ AVLMapIndexed V l u h
→ ∃ 𝜆 i → AVLMapIndexed V l u (i ⊕ h)

insertWith k f leaf = 1# , node (k , f nothing) leaf leaf ~0
insertWith k f (node p lt rt bal) with compare k (proj1 p)

... | tri< k<p _ _ = join𝑙+ p (insertWith k f {{p≤u = [ k<p ]𝑅}} lt) rt bal

... | tri≈ _ refl _ = 0# , node (k , f (just (proj2 p))) lt rt bal

... | tri> _ _ p<k = join𝑟+ p lt (insertWith k f {{[ p<k ]𝑅}} rt) bal

insert : {@0 l u : Key+} {@0 h : ℕ} (kv : Key × V )
{{@erased l≤p : l <+ [ (proj1 kv) ]}}
{{@erased p≤u : [ (proj1 kv) ] <+ u}}
→ AVLMapIndexed V l u h
→ ∃ 𝜆 i → AVLMapIndexed V l u (i ⊕ h)

insert (k , v) m = insertWith k (𝜆 _ → v) m

delete : ∀ {@0 l u : Key+} {@0 h : ℕ} (k : Key)
{{@erased l≤p : l <+ [ k ]}} {{@erased p≤u : [ k ] <+ u}}
→ AVLMapIndexed V l u h
→ ∃ 𝜆 i → AVLMapIndexed V l u pred[ i ⊕ h ]

delete k leaf = 0# , leaf
delete k (node p lt rt bal) with compare k (proj1 p)
... | tri< k<p _ _ = join𝑙− p (delete k {{p≤u = [ k<p ]𝑅}} lt) rt bal
... | tri≈ _ refl _ = join lt rt bal
... | tri> _ _ p<k = join𝑟− p lt (delete k {{[ p<k ]𝑅}} rt) bal

lookup : ∀ {@0 l u : Key+} {@0 h : ℕ}
→ AVLMapIndexed V l u h
→ Key
→ Maybe V

lookup leaf k = nothing
lookup (node p lt rt bal) k with compare k (proj1 p)
... | tri< k<p _ _ = lookup lt k
... | tri≈ _ refl _ = just (proj2 p)
... | tri> _ _ p<k = lookup rt k

B
Simply Typed Lambda Calculus

Interpreter

B.1 Definitions and Typing Rules

𝜏 ∶∶= ℕ | () | 𝜏 → 𝜏
𝑒 ∶∶= 𝑛 | () | 𝑥 | 𝑒1 + 𝑒2 | 𝜆𝑥.𝑒 | 𝑒1 𝑒2
𝑣 ∶∶= 𝑛 | () | 𝜆𝑥.𝑒

(B.1)

Figure B.1: Types, Expressions, and Values for the simply typed lambda calculus.

Nat Γ ⊢ 𝑛 ∶ ℕ Unit Γ ⊢ () ∶ ()

Var 𝑥 ↦ 𝜏 ∈ Γ
Γ ⊢ 𝑥 ∶ 𝜏 Add

Γ ⊢ 𝑒1 ∶ ℕ
Γ ⊢ 𝑒2 ∶ ℕ

Γ ⊢ 𝑒1 + 𝑒2 ∶ ℕ

Abs
(𝑥, 𝜏1) ∶∶ Γ ⊢ 𝑒 ∶ 𝜏2
Γ ⊢ 𝜆𝑥.𝑒 ∶ 𝜏1 → 𝜏2

App

Γ ⊢ 𝑒1 ∶ 𝜏1 → 𝜏2
Γ ⊢ 𝑒2 ∶ 𝜏1
Γ ⊢ 𝑒1𝑒2 ∶ 𝜏2

(B.2)

Figure B.2: Typing rules for the simply typed lambda calculus.

B.2 Agda Implementation

data Unit : Set where
unit : Unit

Var : Set
Var = ℕ

data Type : Set where
nat : Type
unit : Type
_=>_ : Type → Type → Type

[[_]] : Type → Set
[[ nat ]] = ℕ
[[ unit ]] = Unit
[[ 𝜏 => 𝜏 ’ ]] = [[ 𝜏 ]] → [[ 𝜏 ’ ]]

private
variable

Ctx : Set

module Interpreter (bMap : BMap {K = ℕ} {Type} Ctx)
(cAll : CorrectAll {ℓ𝑎 = 0ℓ} {ℕ} {Type} Ctx bMap) where

open Map.BasicMap.BMap bMap public
open Map.CorrectAll.CorrectAll cAll public

Env : Ctx → Set
Env c = All (𝜆 (_ , 𝜏) → [[ 𝜏 ]]) c

data _⊢_ : Ctx → Type → Set where
T-Int : ∀ {Γ : Ctx}

→ ℕ
------------

→ Γ ⊢ nat

T-Add : ∀ {Γ : Ctx}
→ Γ ⊢ nat
→ Γ ⊢ nat

------------
→ Γ ⊢ nat

T-Unit : ∀ {Γ : Ctx}
-------------

→ Γ ⊢ unit

T-Var : ∀ {Γ : Ctx} {𝜏 : Type} {x : Var}
→ x ↦ 𝜏 ∈ Γ

--------------------------
→ Γ ⊢ 𝜏

T-Abs : ∀ {Γ : Ctx} {x : Var} {𝜏1 𝜏2 : Type}
→ (insert x 𝜏1 Γ) ⊢ 𝜏2

------------------------------------
→ Γ ⊢ (𝜏1 => 𝜏2)

T-App : ∀ {Γ : Ctx} {𝜏1 𝜏2 : Type}
→ Γ ⊢ (𝜏1 => 𝜏2)
→ Γ ⊢ 𝜏1

--------------------------
→ Γ ⊢ 𝜏2

translate : ∀ {Γ : Ctx} {𝜏 : Type} → Env Γ → Γ ⊢ 𝜏 → [[ 𝜏 ]]
translate _ (T-Int n) = n
translate env (T-Add e1 e2) = translate env e1 + translate env e2
translate env T-Unit = unit
translate env (T-Var {x = x’} prf ) = allLookup {k = x’} prf env
translate env (T-Abs {x = x} e) e’ = translate (allInsert {k = x} e’ env) e
translate env (T-App e1 e2) = translate env e1 (translate env e2)

Evaluation Figures

Figure C.1: Elapsed time for inserting values in the range 0..𝑛 into an empty map.

Figure C.2: Times for merging an empty map with a map containg 1000, 10,000,
or 100,000 elements using union.

Figure D.1: Maximum residency for inserting values in the range 0..𝑛 into an
empty map.

Figure D.2: Total memory usage for inserting values in the range 0..𝑛 into an
empty map.

Figure D.4: Total memory usage for for union where 𝑛 is fixed with 100 000
elements.

elements.

